
The Joy of Joysticks 
By Jon Williams 
 
For Nuts & Volts Magazine, Column 8, September 2010 
 
 
I think it’s fair to say that the joystick may be the first user input device connected to a computer that took 
us beyond the keyboard – games are far more fun with a stick in our hand than hunting for keys, right?  
The first joysticks, of course, were simply made up of switches pressed by a plate connected to the stick.  
Then came analog  joysticks which were built up with two potentiometers mechanically linked at a 90-
degree angle, one pot for each axis.  The original PC joysticks we easy to connect to microcontrollers 
without hacking.  With the proliferation of USB ports, though, analog joysticks changed their interface and 
are no longer microcontroller friendly.  Darn.... 
 

Game On! 
If you’ve played any video game console in the last 10 years you’ve no doubt had the opportunity to place 
your thumbs on the controller’s mini joysticks and go crazy with your friends.  Until recently, those mini 
joysticks, with their cute, mushroom-shaped caps weren’t very easy to come by and, when you could find 
them, they were pricey – a few years ago I paid about $10 each. 
 
Well, that’s not the case anymore.  Parallax offers a version (#27800) that we can plug into solderless 
breadboards and Nick over at Gadget Gangster is offering them as raw parts that we can build into 
projects.  So the hardware this month is a Propeller Platform module that holds two of those mini 
joysticks, an ADC to read them, and some other bits that we can use in a variety of projects.  Figure 1 
shows my completed prototype. 
 

 
Figure 1 

Using the MCP3204 
The mini joysticks are in fact two 10K potentiometers that are mechanically linked together such that one 
changes with X-axis movement, the other with Y-axis movement.  With two joysticks we’ll need a four-
channel ADC.  We could use an ADC0834 but the MCP3204 seems to be popular amongst Propeller 
users so I thought I’d give it a whirl. 
 
Figure 2 shows the connections between the joysticks and the MCP3204, a four-channel, 12-bit ADC.  
The MCP3204 has four pins that form the communication buss to the host microcontroller, but the 
addition of a 2.2K resistor allows us to cut that down to three.  The resistor protects the Propeller pin from 
a short circuit when DOUT is an output, the IO pin is an output, and the two are in opposite states (one 
high, one low). 
 

 
Figure 2 

 
The process of reading a channel value from the MCP3204 requires these steps: 
 

1. take the /CS line low 
2. output a 5-bit configuration word 
3. read 13 bits from the ADC (null plus 12 bits) 
4. return the /CS line high to deselect 

 
The MCP3204 will handle very high-speed transactions and we could write a driver in Assembly but, 
really, there’s no need for most applications.  I’ve created a simple object, written completely in Spin, that 
will handle reading any of the channels in the two modes that are supported (single-ended and 
differential). 
 
The MCP3204 object has methods to setup the IO pins used by the circuit and two variants that will read 
the value of a channel.  The init() method takes care of the IO pins and places the ‘3204 in the 
deselected state: 

pub init(cspin, clkpin, diopin) 
 
  cs := cspin 
  outa[cs] := 1 
  dira[cs] := 1 
 
  clk := clkpin 
  outa[clk] := 0 
  dira[clk] := 1 
 
  dio := diopin 
 
As you can see, we pass the chip select pin, the clock pin, and the data IO pin to the init() method.  The 
/CS pin is set to output and high to deselect the ‘3204 while the clock pin is set to output and low; the data 
IO pin is left in input mode until we need it. 
 
Astute readers (that would be you!) will ask, “Why do we need to pass pin numbers when they won’t 
change in a program?  Can’t we just set them as constants?”  We could, but it’s a bad idea for reusable 
objects like this.  If we used constants for the pin numbers we’d be forced to either, 1) always use the 
same pin numbers in every design (impractical), or 2) keep multiple copies of the object with different pin 
numbers which is also impractical when it comes to maintenance and upgrades.  So, the lesson here is 
this: when designing a Propeller object keep it concise, yet flexible enough to be re-used in any of your 
designs.  This will make your life simpler and those downloading your code from the Object Exchange 
(obex.parallax.com) will thank you. 
 
There are two variations of read methods and, in fact, one calls the other.  The reason for two methods is 
to accommodate different programming styles.  The first variation expects the channel number (0 to 3) 
and the mode (1 for single-ended, 0 for differential). 
 
pub read(ch, m) | mux 
 
  mux := %1_0000 | ((m & 1) << 3) { 
         } | (ch & %11) 
  return readx(mux) 
 
The channel and mode settings are used to create a 5-bit configuration word with masking and bit 
shifting; this value will passed to the readx() method that actually does the heavy lifting.   
 
pub readx(mux) | level 
 
  outa[cs] := 0 
  dira[dio] := 1 
 
  mux <<= constant(32-5) 
  repeat 5 
    outa[dio] := (mux <-= 1) & 1 
    outa[clk] := 1 
    outa[clk] := 0 
 
  dira[dio] := 0 
  level := 0 
  repeat 13 
    outa[clk] := 1 
    outa[clk] := 0 
    level := (level << 1) | ina[dio] 
 
  outa[cs] := 1 
 
  return (level & $FFF) 

If this code looks a little bit familiar, well, it should.  It’s really a combination of code that simulates the 
PBASIC SHIFTOUT and SHIFTIN instructions; we used code identical the latter portion when reading the 
74x165 shift-register on the encoder board (Nuts & Volts, May 2010).  And just a note on my method 
naming convention: when I have two methods that are similar, and one is a simplified interface into the 
other, the method that really does the work tends to get appended with “x” (for explicit). 
 
We start by taking the /CS line low to activate the ‘3204.  Remember that the init() method made this pin 
an output so all we have to do is set the level by writing to the outa[] register.  The next step is to put 
the DIO line into output mode so that we can write the configuration word to the ‘3204. 
 
The configuration word is shifted out MSB first.  The cleanest way to do this in Spin is to move the MSB of 
the configuration word into bit 31; we do this with a right shift (of 27 bits).  Next, we drop into a repeat loop 
that starts by rotating bit 31 into bit 0; bit 0 is then moved to the DIO pin and the clock pulsed high then 
low to latch the bit into the ‘3204. 
 
To receive the channel value from the ADC we switch the DIO line to input mode, clear the workspace 
variable, and then drop into a repeat loop that shifts the bits in.  As you can see the loop runs 13 times; 
the reason for this is that the ‘3204 outputs a null bit before the first bit (MSB) of the channel reading. 
 
To shift the ADC bits in the clock line is pulsed high, then low; we’ll sample the line after the clock pulse 
(MSBPOST).  The bits arrive MSB first so the work value is shifted left before the DIO line is sampled and 
copied into bit 0. 
 
Finally, the /CS line is taken back high to deselect the ADC and we can return the value to the calling 
program.  Note that while this code is setup for the ‘3204 a very simple change will allow it to work with 
the MCP3208 as well; both devices use a 5-bit configuration word. 
 
Okay, then, let’s put it to use.  I’m a very big believer in creating hardware test programs before moving 
on to the actual application (more on this later), and part of the test program for the board I made will read 
and display the analog joystick inputs.  Figure 3 shows the output of this code in PST. 
 

 
Figure 3 

Do you see a problem?  These readings are with the joysticks at neutral, and in a perfect world all would 
read 2048 as this is the half-way point for a 12-bit ADC.  One of my desires is to control servos with the 
joysticks so a little extra effort will be required to make this work cleanly. 
 
The first thing to do is nullify the neutral position values; the autocal() method does this: 
 
pub autocal | idx, raw 
 
  repeat idx from 0 to 3 
    raw := adc.read(idx, adc#SE) >> 4 
    ofs[idx] := JOY_CNTR - raw  
 
There is actually two things happening here: the channel is read (in single-ended mode) and then shifted 
right by four bits to reduce the ADC resolution down to eight bits; the pots we’re using are not precision 
units and 12 bits is excessive.  By reducing the resolution we can nearly eliminate the LSB “bouncing” of 
the input readings.  The second step is to save the offset from expected center (128 for eight bits) in an 
array that can be applied to future readings. 
 
So let’s say that we want to control four servos with the headers on the board; the final step would be to 
convert the joystick channel reading to the range required by the servo driver; this value will usually be 
expressed in microseconds, and 1000 to 2000 is the standard range.  The joy2servo() method takes an 
eight-bit joystick reading and, using algebra, converts it to a standard servo position value. 
 
pub joy2servo(jpos) | delta, spos 
 
  jpos  := JOY_MIN #> jpos <# JOY_MAX 
 
  delta := (jpos - JOY_MIN) * 1000 { 
            } / (JOY_MAX - JOY_MIN) 
 
  spos  := ((SVO_MAX - SVO_MIN) { 
            } * delta / 1000) + SVO_MIN 
 
  return spos 
 
The first step ensures that the input value in jpos is legal, that is, it fits within the eight-bit range we should 
expect from the joystick.  With the variations in the pots I decided to truncate the readings by 15 on each 
end, so jpos will fall between 15 (JOY_MIN) and 240 (JOY_MAX).  
 
The next step converts jpos to a percentage.  For example, if the joystick is in the neutral position (jpos = 
128) then delta will be 50%.  Since we’re dealing with integers the percentage is multiplied by 1000 to 
prevent rounding errors. 
 
The final step translates the input delta to a servo position value, as defined by the minimum and 
maximum servo position constants.  As you can see, the calculated position delta is applied to the span of 
the servo range and then divided by 1000 to get to the value within the span; this is added to the 
minimum value for the range to get the proper servo position value.  Figure 4 shows the output from 
jm_joystick_servo_demo.spin.  As you can see, the calibration has taken care of getting the neutral 
position where it needs to be, and the center value for a typical servo is properly calculated.  The program 
includes a mini servo driver which allows us to control four servos from the joysticks. 

Figure 4 

 

Pulse Position Magic 
A couple years ago I was visiting my friend Dan at a Hollywood special effects shop, and he showed me a 
box of VEX transmitters that his company had purchased from All Electronics when the price was so low 
(about $30 – and you can still find these units on eBay).  Dan was gutting the joysticks from the 
transmitters for use in animatronics controls, but for many projects, there’s no need to do this. 
 
Most RC transmitters, including the one from VEX, have a trainer connection that emits a pulse stream 
that defines the position data of the joysticks and buttons.  Figure 5 shows the stream from the six-
channel VEX transmitter when connected to a microcontroller.  The actual output of the VEX is open-
collector, so the input to the micro needs a pull-up.   
 
In order to provide the greatest flexibility I use the circuit shown in Figure 6 on the project board.  Two 
jumpers let me configure the 3-pin PPM port; I can provide power to the device (e.g., a VEX receiver) and 
configure the input as pulled-up or pulled down as required.  The purpose of the 2.2K series resistor is to 
limit the current into the Propeller IO pin for those devices that provide a driven output (5V max). 

Figure 5 
 
 

 
Figure 6 

The VEX stream begins with a nine-millisecond sync pulse and is followed by six channel pulses.  All of 
the pulses are separated by an idle period of about 390 microseconds.  The value for a channel is 
determined by measuring the period from one falling edge, through the idle period, to the next falling 
edge.  In the past I’ve measured pulses with the Propeller counters, but this time we’ll need to measure 
the low and high portions of the signal, and we’ll want to ensure that a shorted input or disconnected 
cable does not cause the values into the system to become corrupted. 
 
The only way to measure the PPM pulse stream with precision is to use Assembly code.  I’ll admit that it 
is a little involved, but really not too bad when broken down piece-by-piece.   Let’s jump in. 
 
At the very top we’re going to start by writing a flag indicating there is no detected PPM stream, then we’ll 
monitor the input for a valid sync pulse. 
 
vexin     mov     dira, #0 
          wrlong  NO, flagpntr 
 
findsync  mov     pwidth, #0 
          mov     timer, US_001 
          waitpne pmask, pmask 
          add     timer, cnt 
:loop     waitcnt timer, US_001 
          add     pwidth, #1 
          test    pmask, ina    wz 
  if_z    jmp     #:loop 
          cmp     pwidth, SYNC  wc, wz 
  if_b    jmp     #vexin 
 
At findsync the pulse width variable is cleared and a timer is setup for one-microsecond delays with 
waitcnt.  Then we wait for the input to be low using waitpne.  Note that this is the only place in the code 
where we will use a pin wait instruction; from this point forward we will manually scan the input so that we 
can do it on a timed interval (1us). 
 
Once a low on the input is detected the timer is started and drops into a loop, incrementing the pulse 
width measurement every microsecond until the line goes back high.  We scan the PPM input using the 
test instruction with a mask for the input pin (pmask) and the pin inputs register (ina).  The test 
instruction works exactly like the and instruction but it doesn’t affect the variable in the destination field.  
The Z flag is updated with the state of the PPM input and as long as remains true (Z = line is low) the 
code will loop and increment the pulse measurement variable.  When the input goes high (during the 
inter-pulse idle period) the Z flag will be cleared and the code drops into a comparison to check the width 
of the pulse.  If the pulse we just measured is not a valid sync pulse (for example, we started the code in 
the middle of an output stream) then the code jumps back to the top and tries again. 
 
The next section of code monitors and measures the idle period between the sync pulse and the first 
channel pulse.  The reason we bother measuring is to prevent a disconnected device from locking the 
program (since we’re waiting for a low-going leading edge for the channel measurement). 
 
          mov     pwidth, #0            
chwait    test    pmask, ina    wz       
  if_z    jmp     #getch1               
          waitcnt timer, US_001         
          add     pwidth, #1            
          cmp     pwidth, #425  wc, wz   
  if_b    jmp     #chwait                          
          jmp     #vexin   
 
As before, the pulse width variable is cleared and the code drops into a small loop that tests the state of 
the PPM line.  When the line is idle the Z flag will be cleared and we drop through the first jump into a 
one-microsecond delay.  The idle pulse width is incremented and a comparison is done to ensure that we 

haven’t lost the signal.  If the PPM signal is lost the input will remain high and code is redirected back to 
the top of the program where we’ll wait for reconnection and a valid sync pulse. 
 
When things are working as they should, the PPM input will drop after about 390 microseconds – this is 
where we start the channel position measurement.   This code looks – and is – very easy; it’s actually just 
a call to a subroutine and a check on an error flag (stored in C) that can be set by that routine. 
 
getch1    call    #getch 
  if_c    jmp     #vexin 
          mov     ch1us, pwidth 
 
As you can see, this section calls a subroutine called getch which performs the actual channel 
measurement and will set the C flag if a measurement error is detected.  After returning from the 
measurement routine we check the C flag and if it’s set, jump right back to the very top of the program 
where the PPM stream detection flag is set to NO (false).  This flag can be used by our application to 
determine the presence or absence of the PPM stream and take some action based on that status.  
Assuming a good measurement, though, we save the channel timing in a holding register and then do the 
same operation for the rest of the channels. 
 
The hard work is performed by the getch subroutine.   
 
getch     mov     pwidth, #0 
:looplo   waitcnt timer, US_001 
          add     pwidth, #1 
          cmp     pwidth, LO_MAX  wc, wz 
  if_a    jmp     #badch 
          test    pmask, ina      wz 
  if_z    jmp     #:looplo 
 
:loophi   waitcnt timer, US_001 
          add     pwidth, #1 
          cmp     pwidth, CH_MAX  wc, wz 
  if_a    jmp     #badch 
          test    pmask, ina      wz 
  if_nz   jmp     #:loophi 
 
goodch    cmp     pwidth, CH_MIN  wc, wz 
          jmp     #getch_ret 
 
badch     test    $, #1           wc 
getch_ret ret 
 
Yeah, this does look a little bit hairy, but as you examine it closely you’ll find it’s divided into two sections: 
the first to measure the low portion of the period, the second to measure the idle period which completes 
the channel timing. 
 
Again, we start by clearing the timing measurement variable (pwidth) and then doing a one microsecond 
delay.  The pulse timing is incremented and then compared to a constant that defines the longest valid 
low period.  We do this so that a short on the PPM input can be detected.  If this happens the C flag gets 
set and we abort back to the calling code. 
 
This usually won’t happen, though, and we continue timing until the PPM line goes back high for the idle 
period.  Now we time this while waiting for the line to go back low (the start of the next channel), again, 
testing against an established limit.  The test during the high period will detect a disconnection of the VEX 
device from the circuit.  When we return to the calling section the variable called pwidth will contain the 
channel value and when all is well the C flag will be cleared. 

After all of the channels have been measured we can write the values to the hub for access by the 
application. 
 
report    mov     pntr, pospntr 
          wrword  ch1us, pntr 
          add     pntr, #2 
          wrword  ch2us, pntr 
          add     pntr, #2 
          wrword  ch3us, pntr 
          add     pntr, #2 
          wrword  ch4us, pntr 
          add     pntr, #2 
          wrword  ch5us, pntr 
          add     pntr, #2 
          wrword  ch6us, pntr 
 
          wrlong  YES, flagpntr 
          jmp     #findsync 
 
This is pretty straightforward: we setup a pointer to the hub array (of words) that holds the channel values 
and then transfer the readings from the cog.  At the end we write YES (true) to the flag that indicates PPM 
status to alert the application that the stream is present and the position values are fresh. 
 
Figure 7 shows the output from a test program that reads and displays the PPM stream status and 
channel values. 
 

 
 

Figure 7 

Testing 1, 2, 3.... 
I often see posts in the Propeller support forum that go something like this: “I built my hardware, I wrote 
my software – it doesn't work.  What do I do?” 
 
Test. Test.  Test.  The good news is that the Propeller architecture helps us.  Earlier this year my friend 
and I created and released a Propeller-based WAV audio player (the EFX-TEK AP-16+) that is designed 
for museums, theme parks, and attraction-based industries like Halloween.  It’s a beast.  Because of the 
variability of the user base it has lots of flexibility in its design which means lots of IO and a fairly-tricky 
PCB. 
 
While John (who also worked for Parallax) was creating the PCB for the first prototype, I wrote a test 
program to check each section of the board – this was well before a single line of application code was 
written.  Of course, that’s a bit of a fib, isn’t it?  You see, a large portion of the code used in testing moved 
right into the final application.  In fact, as I refined the test program I was refining portions of the 
application code. 
 
The ability to program the Propeller through a serial port means that we can create a self-hosted test 
program that uses a simple terminal.   As an example for you to play with, I created a test program for the 
this month’s project board.  When that program starts it presents a menu using PST (see Figure 8) that 
lets us access/test each section of the board.   
 

 
 

Figure 8 
Here’s the main program loop: 
 
pub main | c 
  term.start(RX, TX, %0000, 115_200) 
  rgled.initx(R_LED, G_LED, 50, OFF) 
  adc.init(CS, CLK, DIO) 
  ppm.init(PPM_IN, 1500) 
  eeprom.init(SCL, SDA) 
 
  pause(1) 

  repeat 
    term.tx(CLS) 
    term.str(@Menu) 
    term.rxflush 
    c := term.rx 
    case c 
      "1" : showdigins 
      "2" : demoled 
      "3" : readjoysticks 
      "4" : readppm 
      "5" : testeeprom 
 
Once the objects used by the board are initialized we drop into a loop that displays the main menu and 
waits for input.  Based on that input the program will jump to one of the test routines. 
 
The menu is defined as a string and is stored in a dat section of the program like this: 
 
Menu    byte    "Waldo-4x Hardware Checkout", CR 
        byte    "==========================", CR 
        byte    CR 
        byte    "[1] Digital Inputs        ", CR 
        byte    "[2] Bi-Color LED          ", CR 
        byte    "[3] Read Joysticks        ", CR 
        byte    "[4] Read PPM Stream       ", CR 
        byte    "[5] External EEPROM       ", CR 
        byte    0 
 
Note that the menu string can be spread across lines which allows us to “see” it in the listing.  Note, too, 
that it must be terminated with a zero; we’re actually defining the menu as a single z-string and if we omit 
this the display could end up a mess.  The listing above shows that displaying the menu is simply a 
matter of passing the address (@) of the menu data to the str() method of the terminal object.  
 
We can have sub-menus in the test sections, or run code in a simple loop; like this code that displays the 
values from the VEX PPM stream. 
 
pub readppm | idx, c 
 
  term.rxflush 
  term.tx(CLS) 
  term.str(string("PPM Stream", CR, { 
           } "-- press X to quit")) 
 
  repeat 
    term.str(string(HOME, LF, LF, LF)) 
    term.str(string("Active   ")) 
    if (ppm.status == false) 
      term.str(string("No  ", CR)) 
    else 
      term.str(string("Yes ", CR, CR)) 
      repeat idx from 1 to 6 
        term.str(string("Ch")) 
        term.tx("0" + idx) 
        term.str(string("      ")) 
        term.dec(ppm.read(idx)) 
        term.tx(CLREOL) 
        term.tx(CR) 
 
    c := term.rxtime(50) 
    if (ucase(c) == "X") 
      quit 

At the top we clear the terminal RX buffer to remove extraneous keys, remove the main menu, and then 
display a title string for this test (and note on how exit).  The rest of the code is contained in a loop that 
displays the VEX connection status and channel values. 
 
At the bottom you’ll see a neat trick with the rxtime() method.  This method lets us check for input from 
the user, but only for a specified period (in milliseconds).  If no key is pressed in this time the method 
returns with a value of -1 (not a valid key).  We check the return value for “X” and when that’s detected, 
abort back to the main menu.  Until that time, the rxtime() method provides a nice 50-millisecond update 
rate for the display – it’s a 2-for-1 deal. 
 
Study the test program and put it to use with your own projects.  Remember, the time you spend working 
on test code is not wasted because this code checks your hardware (you need this) and can then be 
moved into your final application.  That test program I created for the AP-16+ prototype is now used on 
the other end of the product cycle as a unit tester that is run before we load the application code and ship 
to our customer. 

Coding and Publishing are Difficult! 
My friends at Nuts & Volts do a spectacular job publishing this (and other) magazines for people like you 
and me.  It’s very hard work, taking the frequently wacky style of programmers and putting it into print.  I 
mention this because after my last column a misguided young man (Ah, to be young and stupid....) took 
me to task for not providing comments in my code (as it appears in print). 
 
Of course, I comment my all of my code, in fact, like most professional programmers, I take pride in the 
quality of my work and the information that goes into it.  That said, the two-column format that is preferred 
by Nuts & Volts and other publications just doesn’t lend itself to long program lines so, of necessity, I strip 
out the inline comments for the code that is embedded in the article text. 
 
I realize that this stuff can at times get a little heavy, especially the Assembly sections.  So those of you 
really wanting to study the code as you read my column (or others, for that matter) should open the code 
—that can be downloaded from Nuts & Volts—in your programming editor as you’re reading the 
magazine.  Studying the code comments along with the article text will really help solidify your 
understanding of the project. 
 
Until next time, then, keep spinning and winning with the Propeller. 
 

Resources 
 
Jon “JonnyMac” Williams 
jwilliams@efx-tek.com 
 
Parallax, Inc. 
www.parallax.com 
 
Gadget Gangster 
www.gadgetgangster.com 
– Propeller Platform kits and accessories 

Bill of Materials 
 

C1-C4 0.01uF Mouser 80-C315C103M5U 
C5 220 Mouser 140-L10V220-RC 
JP1-JP4 0.1 M-STRT Mouser 517-6111TG * 
LED1 G/R bi-color Mouser 604-WP937EGW 
Q1 2N3904 Mouser 610-2N3904 
R1-R4 100 Mouser 291-100-RC 
R5-R6 470 Mouser 291-470-RC 
R7-R8 2.2K Mouser 291-2.2K-RC 
R9 10K Mouser 291-10K-RC 
RN1 10K Mouser 81-RGLD8X103J 
RX1-RX2 10K Dual Gadget Gangster RKJXK122000D 
SW1 DIP-6x Mouser 611-BD06 
SW2 6mm NO Mouser 653-B3F-1022 
S-U1 14-pin DIP Mouser 571-1-390261-3 
U1 MCP3204 Mouser 579-MCP3204-CI/P 
X1-X10 0.1 M-STRT Mouser 517-6111TG * 
PCB  ExpressPCB 
Parts Kit  Gadget Gangster 

* The 517-6111TG is a 40-pin component; only three are required for the project (split as needed). 


