
Loving LEDs Again 
By Jon Williams 
 
For Nuts & Volts Magazine, Column 1, July 2009 
 

I often tell those friends that are just getting into microcontrollers that if they can learn to control an LED 
they can learn to control just about anything.  Now, most of those friends are prop builders and, for the 
most part, are just looking for simple on and off control; a blinking LED easily becomes a pneumatic 
piston raising and lowering some element of a prop or special effect.  I recently used a bicolor LED in an 
SX-based product design and it was a big hit – the product, and the use of the LED.  With a simple, two-
lead, bi-color (green and red) LED I was able to communicate the state of the machine and our customers 
loved it.  Program space, however, prevented me from taking the LED control code as far as I wanted to.  
Well, with a whole cog at my disposal that is no longer an issue and I am loving LEDs again! 
 
The fact of the matter is that we humans are reasonably trainable (if you’re a girlfriend, housewife, or 
mother you’re probably shaking your head in disagreement at the moment, but bear with me).  This 
statement is especially true when new training is built on established skills or information.  For example, 
we often use the color green for “go” or “okay”; red means “stop”; yellow means “caution” – of course, all 
of this depends on context.  In recording devices, green usually means “play” and red means “record.”  As 
product designers it is in our best interest to take advantage of previously-established training so that our 
customers can adapt to the new product more quickly. 
 
The product I mentioned earlier is a digital recorder and playback device designed for prop and 
animatronics control.  When armed for playback the mode LED is green; when playing a recorded 
sequence the LED flashes green; I used motion (blinking) to indicate that the device is active.  When a 
channel is armed for recording the mode LED is red; when recording is in progess the LED flashes red.  
I’m sure this seems terribly obvious as you’re reading it now but how many products have you dealt with 
that didn’t use obvious choices like these?   
 

Building an LED Coprocessor 
Since the early days of BASIC Stamps we’ve become accustomed to adding application-specific co-
processors to enhance a project; things like servo controllers and serial LCDs – they’re so common now 
that we take these things for granted.  One way to think of the Propeller is a 32-bit processor with seven 
built-in coprocessor engines that we can do with as we please.  Neat, huh?   
 
In my SX project the LED control code was part of the ISR so it had to be very trim and this limited the 
features I could include to color and basic on-off blinking.  What I really want is the ability to control color, 
do colorized blinking, and have brightness control of the LED.  That’s what we’re going to do with the 
Propeller.  We’ll start simply and build as we go until we have the full-featured driver. 

The hardware for this project is simple: a resistor and bicolor LED that you probably have in your parts 
drawer.  You can solder these together for the Propeller Platform or plug them in if you’re using the 
Propeller Demo Board or PPDB. 
 

 
 

Figure 1: Bicolor LED Schematic 
 
 

Three Colors from Two 
The two-lead, bicolor LED actually has two LED chips inside: green and red, wired back-to-back.  This 
arrangement allows current to flow through just one element at a time.  It turns out that if we reverse 
current flow quickly through the bicolor LED it can appear yellow, giving us a third color and, in total, four 
states: off, green, red, yellow. 
 
Here’s the rub: green and red LEDs are constructed from different materials and have different forward 
voltages.  On my LED the green chip had a forward voltage of 1.75v and the red chip had a forward 
voltage of 1.56v.  With a 220-ohm resistor and a 3.3v output form the Propeller the green LED would get 
about 7mA and the red LED would get about 8mA – it’s enough to make a difference and you can clearly 
see it when reading the forward voltage with a multimeter. 
 
So what does this mean?  Well, it means that we cannot get yellow by simply reversing the current 
direction every other cycle.  What we’re going to do is construct the code such that we can easily control 
the green-to-red cycles ratio to get the best approximation of yellow. 
 
Okay, then, let’s jump into the code.  You’ll remember from last time that Propeller assembly code 
(PASM) segments are defined in a DAT section of a Spin program.  We’ll start with the interface which 
lets us define the pin to use for the green anode (red will be green plus one) and the color state for the 
LED. 
 
DAT 
                org     0 
 
leddriver       mov     tmp1, par  
                rdlong  tmp2, tmp1 
                mov     gMask, #1  
                shl     gMask, tmp2 
                or      dira, gMask  
 
                mov     rMask, gMask  
                shl     rMask, #1  
                or      dira, rMask  
                         
                add     tmp1, #4 
                mov     colorAddr, tmp1  
 
 
The driver starts by copying the cog’s par (parameter) register to tmp1; what we’re going to pass in par is 
the address (using @) of the variable that holds the green anode pin number (this is the starting address 
of all the parameters we will pass to the driver).  With rdlong we can read the pin number from the hub 
RAM into the cog variable called tmp2.  What we want to do is create a pin mask for the green pin; to do 

that we’ll load gMask with one and then shift left by the pin number in tmp2.  To make that pin an output 
we’ll or the pin mask onto the dira register as dira controls the pins’ IO direction (one bit = output, zero bit 
= input). 
 
In the program we’re going to specify that the red anode pin is the next pin up from green, so creating the 
red mask is easy: we copy the gMask into rMask and then shift rMask left by one bit.  As with the green 
pin we make the red anode pin an output by or-ing the pin mask onto dira. 
 
At this point tmp1 is equal to par which is the address of the green pin variable.  The second parameter 
for the driver is the color mode.  To get the hub address of the color mode variable we can add four (four 
bytes per long) to tmp1 and copy that into colorAddr; we’ll use this variable in the program to update the 
LED color. 
 
With the driver setup [mostly] out of the way we can get into the main loop.  In this version of the driver 
we don’t need any specific loop timing but it doesn’t hurt to add it in to support timing functions to come 
later.  Loop timing in assembly is usually a nightmare of cycle counting but the Propeller relieves us of 
that nightmare by including an instruction called waitcnt which takes care of the dirty work. 
 
Remember that all cogs are driven by a common clock and have access to the system counter.  In both 
Spin and PASM the waitcnt instructions will wait for the system counter to reach a specific value – this 
makes precise timing very simple.  We start by defining a timer variable and loading it with the number of 
counter ticks we want to wait.  As the system timer is almost always running at 80,000,000 ticks per 
second the delay time will usually be a significant number.  Once we’ve loaded the variable with the delay 
ticks synchronize it by adding the current value of the system counter. 
 
starttimer      mov     loopTimer, us010 
                add     loopTimer, cnt  
 
Here we’re loading the variable called loopTimer with the value stored in us010; this is a predefined value 
that you’ll see at the end of the assembly listing. 
 
us010           long    80000000 / 1000000 * 10 
 
At first this may seem odd because it looks like we’re doing long division and multiplication in an 
assembly program.  We’re not.  All PASM programs start as a DAT section in a Spin program and when 
launched with cognew are moved to the cog RAM.   Since we started in Spin the compiler will resolve 
this expression prior to downloading the program. 
 
We could, of course, have stated the value directly but here is a case for writing code that explains itself.  
In this case we take the number of system counter ticks in one second and divide by one million to get the 
number of ticks in one microsecond.  Now we multiply by 10 to get the number of ticks in 10 
microseconds.  This may at first seem like an arbitrary value but it isn’t.  More on that later. 
 
Let’s drop into the main loop.  Its purpose is to read the color setting from the hub RAM and then jump to 
the handler that sets the LED to that color. 
 
ledloop         rdlong  color, colorAddr  
                         
setled          mov     tmp1, color  
                max     tmp1, #3  
                add     tmp1, #jmpcolor0 
                jmp     tmp1 
 
jmpcolor0       jmp     #ledoff 
jmpcolor1       jmp     #ledgreen 
jmpcolor2       jmp     #ledyellow  
jmpcolor3       jmp     #ledred 

The first part is easy with rdlong; this lets us read a value from the hub RAM into a cog variable.  What 
follows is the equivalent of a PBASIC or SX/B BRANCH instruction.  The color value is copied into tmp1.  
Now, tmp1 should be something between zero and three but as good programmers we don’t take 
chances and will do a bit of error trapping; in this case we limit the value of tmp1 to three by using the 
max instruction.   
 
And now for a little tricky stuff: we add the starting location (cog RAM) of a table of jmp instructions to the 
value of tmp1 and then jump to the result.  When tmp1 is zero we will end up jumping to ledoff, when 
tmp1 is three we will end up jumping to ledred.  We could have used cmp (compare) and if_e  jmp with 
each value, but this version of the code is a little cleaner, especially with long tables.  Note that I’ve added 
extra labels to the table for clarity. 
 
Okay, lets control the LED – it’s nothing more that setting the direction of current flow (if any) through it. 
 
ledoff          andn    outa, gMask 
                andn    outa, rMask 
                jmp     #loopwait 
 
ledgreen        or      outa, gMask 
                andn    outa, rMask 
                jmp     #loopwait 
 
ledyellow       add     rgCntr, #1 
                cmp     rgCntr, #16     wc 
        if_b    jmp     #ledgreen 
                mov     rgCntr, #0 
 
ledred          andn    outa, gMask 
                or      outa, rMask 
 
loopwait        waitcnt loopTimer, us010 
                jmp     #ledloop    
 
To turn an output pin off we need to set its driver to zero and the andn (and not) instruction – when used 
with a mask – takes care of that for us.  As I stated the last time Propeller owes many of its features to the 
success of the BASIC Stamp and, believe it or not, the &/  (and not) operator is part of PBASIC 1.0, the 
language that runs on the original BASIC Stamp released in 1993! 
 
The andn instruction actually does two things: it inverts the bits of the second value and then ANDs the 
two values together; the result is written back into the first value.  So we take a mask that has one bit set 
that represents the pin we want to affect.  By inverting this mask we now have a value that is all ones 
except for the bit that represents our pin – this bit is set to zero.  By and-ing the outputs register (outa) 
with the inverted value our pin is set to zero and all others preserved.  So, to turn off the LED we use 
andn to write zeroes to the green and red control bits in outa.  With both pins set to the same level no 
current flows and the LED is off.  The code for green and red should be easy to follow: turn one pin on 
and the other off – this falls right into the no-brainer basket.  But what about yellow? 
 
I stated earlier that we have to run the green side of the LED more cycles than the red to create a color 
that approximates yellow.  With my LED I found that 15 cycles of green to one cycle of red looked good.  
So when we jump to ledyellow we’re going increment a cycle counter (rgCntr) and check to see if it has 
reached 16.  While rgCntr is less than 16 we jump to the green handler, when rgCntr hits 16 we drop 
through, reset the counter and then jump to the red handler. 
 
Another thing we have to get used to when using the Propeller is that the C (carry) and Z (zero) flags are 
not automatically updated as with other controllers (e.g., SX).  The upshot is that we can maintain the 
flags through other instructions and not have to worry about them being clobbered; the downside is, of 
course, that we have to remember to tell the assembler to enable them. 

You’ll see wc – called an effect – at the end of the cmp line in the yellow handler; this will write the carry 
flag based on the result of the compare.  The compare instruction works very much like sub except that it 
doesn’t affect the first value.  When the first value is less than the second the carry flag will be set (1).   
 
Do yourself a favor and put a one of those sticky flags on Table 5-2 in the Propeller Manual (v1.0), this 
table lists the aliases for the various combinations of C and Z flags that can be used to conditionally-
control instructions.  One of the aliases for the carry flag being set is if_b which is short for IF BELOW.  
As you can see, we’ve applied this to the jmp that follows the compare to make it conditional.  When the 
value of rgCntr is below 16 then we jump to the green handler, otherwise we drop through, reset rgCntr 
and then jump to the red handler.  The end result is a modulated LED that appears yellow. 
 
Finally, we drop into a waitcnt instruction that will cause the program to pause until the system counter 
matches the value in loopTmr.  Another feature of waitcnt is that it can automatically reload the control 
variable – this makes using it to control loop timing very simple. 
 
Alright, the driver is done so let’s fold it into a Spin program.  You’ll find that most cog-based objects 
include methods called start and stop. 
 
PUB start(p, c) : okay 
 
  stop 
 
  gPin := 0 #> p <# 26  
  ledColor := LED_OFF #> c <# LED_RED  
 
  okay := cog := cognew(@leddriver, @gPin) + 1   
  
PUB stop 
 
  if cog 
    cogstop(cog~ - 1) 
 
Note that the first thing the start method does is call stop; this is important because we don’t want two 
cogs attached to the same object.  The stop method checks to see if the LED driver is already running 
(cog will be non-zero) and terminates it if it is. 
 
The start method expects two parameters: the pin number for the green anode and the initial color state 
of the LED.  Note that use of the #> (minimum) and <# (maximum) operators in the assignments of gPin 
and ledColor, this ensure that we pass legal values to the driver.  You may wonder why the green anode 
is limited to pin 26.  My logic is this: pins 31 and 30 are used for the serial connection to program the 
Propeller, and pins 29 and 28 connect to the EEPROM that stores the program – this leaves pins 27 and 
lower to use.  We limit the green anode pin to 26 because the driver assumes that the red anode is the 
next pin up from the green pin.  Finally, you can see that cognew is passing the address of gPin, not the 
value of gPin.  This is critical because we need to know the address of the parameter(s) we’re passing to 
the driver. 
 
Before getting to the LED demo lets round things out with methods that will make setting the LED color 
simple and virtually self-documenting. 
 
PUB setcolor(c) 
  
  ledColor := LED_OFF #> c <# LED_RED  
 
PUB off 
 
  ledColor := LED_OFF 

PUB green 
 
  ledColor := LED_GRN 
 
PUB yellow 
 
  ledColor := LED_YEL 
   
PUB red 
 
  ledColor := LED_RED 
 
 
For just a little bit of typing – that I’ve already done for you! – we can have code that looks like this: 
 
  myLed.setcolor(LED_GRN) 
  myLed.red 
 
You’ve certainly noticed the use of constants when setting the LED colors.  A nice feature of Spin is that 
we can create a list of enumerated constants with a simple line, so the color constants declaration looks 
like this: 
 
  #0, LED_OFF, LED_GRN, LED_YEL, LED_RED 
 
 
Okay, then, let’s put it work: 
 
PUB main | idx 
 
  start(0, LED_OFF) 
 
  repeat  
    repeat idx from LED_OFF to LED_RED  
      setcolor(idx) 
      waitcnt(MS_001 * 500 + cnt)  
    off  
    waitcnt(MS_001 * 250 + cnt) 
    green 
    waitcnt(MS_001 * 250 + cnt) 
    yellow 
    waitcnt(MS_001 * 250 + cnt) 
    red 
    waitcnt(MS_001 * 250 + cnt) 
    
The code in the main method launches the led driver with the start method specifying pin 0 as the green 
anode and beginning with the LED off.  The first repeat puts the code that follows into an infinite loop.  At 
the top of that is a small loop that runs through the colors using the setcolor method to update the LED.  
After that we use the individual color methods with 250-millisecond delays in between. 
 
You’ll find this code in jw_bicolor.spin.   This is just the starting point – what we want to do is separate 
the LED code so that we can use it in other projects.  Open jw_bicolor_v1_demo.spin (LED demo code) 
and jw_bicolor_v1.spin (object methods and driver) to see how I’ve separated the constituent parts.  To 
use the driver as part of a project we must declare it as an object. 
 
OBJ 
 
  led : "jw_bicolor_v1" 

PUB main | idx 
 
  led.start(0, led#LED_OFF) 
 
  repeat 
    repeat idx from {  
      } led#LED_OFF to led#LED_RED 
          led.setcolor(idx) 
          waitcnt(MS_001 * 500 + cnt) 
       
    led.off  
    waitcnt(MS_001 * 250 + cnt) 
     
    led.green 
    waitcnt(MS_001 * 250 + cnt) 
     
    led.yellow 
    waitcnt(MS_001 * 250 + cnt) 
     
    led.red 
    waitcnt(MS_001 * 250 + cnt) 
 
See how we use the dot notation for object methods and how constants defined inside the object are 
used in the program.  These changes are subtle, and yet very important.  
 

Blink Baby, Blink! 
Okay, color control is nice but we’re talking about an LED coprocessor here, so let’s rev it up!  I envision 
having the need to set an LED to blink at a specific rate and it would be great if it could alternate between 
two colors of my choice (I consider off a color).  Open the updated driver (jw_bicolor_v2.spin) and have 
a look at how we do this. 
 
The first thing to do is add more control variables; we have two colors (one for each phase of the blinking) 
and two timing periods that control the blinking rate and duty cycle.  And, don’t worry, we can still have a 
solid color; all we have to do is set both phases to the same color and the LED will not blink. 
 
Remember that 10-microsecond timing unit I selected earlier?  Here’s why I used that value: if we run a 
10-microseond loop one hundred times it will consume one millisecond, and a millisecond is a convenient 
unit of timing for delays (we’re used to it from PBASIC and SX/B PAUSE).  You know where this is going: 
the timing units for each phase will be specified in milliseconds. 
 
getphases       rdlong  color0, color0Addr 
                rdlong  time0, time0Addr 
                rdlong  color1, color1Addr  
                rdlong  time1, time1Addr 
                add     phaseTimer, #1 
 
runphase        tjnz    phase, #phase1 
 
phase0          mov     color, color0 
                cmp     phaseTimer, time0   wc 
                jmp     #checkphase 
                          
phase1          mov     color, color1 
                cmp     phaseTimer, time1   wc 
 
checkphase       
        if_ae   xor     phase, #1 
        if_ae   mov     phaseTimer, #0  

The top of the 1ms loop is at getphases where we will read the current color and time settings for each 
phase.  After reading the [new] settings a timer for the current phase is incremented and we drop through 
to runphase which is the top of the 10 µs loop. 
 
Since we only have two phases we don’t need a jump table as when selecting color, so the tjnz (test and 
jump if not zero) instruction works perfectly to get us to the phase handlers.  Inside each handler we 
move the phase color to the color control variable and then check the current phase timer against the 
time specified for that phase, using the carry flag to record the result of the comparison.  At checkphase 
we check to see if the current phase time is above or equal to (if_ae) the phase setting and when that is 
true we flip the phase with xor (0 to 1, 1 to 0) and reset the time.  This is a case that demonstrates the 
preservation of the carry flag and its use as a conditional controller with multiple instructions.  Oh, and if 
you’re wondering what happens when the condition fails… the instruction is treated like a nop (it 
consumes four clock cycles but doesn’t do anything). 
 
The color control code doesn’t change so let’s address the end of the loop. 
 
loopwait        waitcnt loopTimer, us010  
                add     loopCntr, #1 
                cmp     loopCntr, #100      wc  
        if_b    jmp     #runphase 
                mov     loopCntr, #0  
                jmp     #getphases  
  
This probably makes a lot of sense by now.  After waiting for the 10us timer to expire we’re going to 
update a loop counter and compare it to 100 – again, 100 x 10us = 1ms.   When the loop counter is less 
than 100 we jump to the inner loop at runphase, otherwise we reset the counter and jump to the top of the 
outer loop at getphases. 

It’s Better to Fade Away…. 
Let’s jump into the driver one more time and add a final feature: brightness control.  As you’d expect this 
is going to be accomplished with PWM (pulse width modulation) of the LED but there is a catch: we’re 
using a bicolor LED and current can be flowing in either direction (or alternating), so how do we modulate 
brightness?  There answer is actually pretty simple: we modulate the direction control pin of one of the 
anodes.  You see, if we make either anode control pin an input there will be no current flow – in any 
direction – through the LED. 
 
Here’s another reason why I went with a 10 µs inner loop: multiplying that by 100 we get a convenient 
one-millisecond timing interval and we also have a counter to use for on-off control of the PWM cycle in 
1% increments.  Let’s look a the code that gets inserted before the loopwait section (you’ll find this in 
jw_bicolor_v3.spin) 
 
pwmctrl         test    level, level        wz 
        if_z    andn    dira, gMask 
        if_z    jmp     #loopwait 
                cmp     loopCntr, level     wc  
        if_b    or      dira, gMask 
        if_ae   andn    dira, gMask 
 
The top of the pwmctrl section first checks the value of level (brightness setting) for zero; when it is we 
will turn off the output (by making the green pin an input) and skip right to loopwait.   When level is 
something greater than zero it is compared to loopCntr and when loopCntr is below the level setting we 
will turn the led on; when loopCntr reaches level the on-cycle is complete and we turn the LED off. 
 
In the end what we get is fixed-frequency, variable duty-cycle PWM that cause the LED to be on for the 
percentage set in level.  So what happens if we set level greater than 100?  Nothing harmful; at 100 or 
higher the LED will be on all the time as the value of loopCntr cycles between 0 and 99.  I like this kind of 

fixed-frequency PWM code because it’s easy to configure.  I recently used it in an RGB LED object and it 
worked really well. 
 
Before I close I want to tell you something about jw_bicolor_v3_demo2.spin.  This program shows how 
we can launch another cog to control the brightness of the LED object in the background; for example, to 
handle automated fades.  The key here is having the LED object reveal the hub address of its brightness 
control variable.  With this information another object can manipulate brightness for advanced control.  
Check it out, it’s pretty neat. 

Propeller Platform Kits! 
Good news, gang!  The response to the Propeller Platform was so positive (one Parallax forum member 
dubbed it “Propellerino”) that my good friend, Ken Gracey, has agreed to make kits available through 
Parallax.  This will certainly simplify building the P/P and save you money over having to buy raw PCBs 
yourself.  Figure 2 shows the final prototype; I really like the way it turned out.  The only change from the 
original design is the regulators.  We changed from the LM108x series to the LM29xx series to simplify 
parts procurement.  My original thought for the high-current regulators was that I could use them for 
powering servos.  Well, having worked with a lot of servo animatronics between the first version and now I 
found that this doesn’t help and for big animatronics an external servo power supply is in order. 
 

 
 

Figure 2: The Propeller Platform 

Wrap Up 
Okay, admit it… when you first started reading this column you groaned, “Come on, Williams, LEDs 
again?... Are you nuts?”  Yeah, I get it, but I think I’ve shown that the lowly LED can become a thing of 
wonder with some cool code in the Propeller chip.  I’m a big believer learning new coding techniques with 
simple hardware and you can’t get much simpler than an LED, right?   
 
Okay, it’s time go experiment because we’ve laid the groundwork for a pretty serious project next time: 
Waldo Part Deux.  Yep, the Waldo project from back in September of 2007 was hugely popular so for its 
two-year anniversary we’re going to recreate it for the Propeller and even add two more channels.  The 
hardware will be a plug-in module for the Propeller Platform that includes mini joysticks and all the 
support circuitry.  In addition to the LED object we just created we’ll also create and use background ADC 
and servo control objects – it’s going to be a lot of fun and a very cool project for use in your holiday 
animatronics displays. 
 
Until next time, then, here’s to spinning and winning with the Propeller. 
 

Bill of Materials 
LED1 bi-color, green/red Mouser 78-TLUV5300 
R1 220 Mouser 299-220-RC 

 


