
Propeller Time 
By Jon Williams 
 
For Nuts & Volts Magazine, Column 7, July 2010 
 
 
A couple months ago I created an encoder object for my friend, Wayne (“the Brain”).  It turns out that 
Wayne’s project is a multi-function timer which I helped him improve a bit through the use of a very simple 
– No Assembly Required – timer object.  The Propeller’s architecture gives us the ability to be fairly 
precise about timing without much effort, and I thought it was time we delve into that a bit.  Timing, that is. 

Slow Down! 
It is somewhat humorous that we all demand faster processors and then turn around and do things to 
slow them down, usually to create delays for blinking LEDs and similar processes.  Of course, using the 
Propeller, we could easily launch a blinking LED cog, but this use of resources is not always required. 
 
Since the BASIC Stamp 1 (released in 1993) most of us have become accustomed to PAUSE.  Very 
handy—so much so that you can find some variant of it in nearly every embedded language going.  
Except Spin, that is....  What gives? 
 
Keep in mind that the Spin interpreter runs in a cog and, therefore, must be very trim.  What this means 
for us – especially those who migrated from some form of embedded BASIC – is that a lot of niceties 
we’ve used in the past just don’t exist in Spin (we discussed this last time when creating a method to 
emulate PBASIC’s SHIFTIN function). 
 
The Propeller does have a method for creating delays but it takes a bit of getting used to and must be 
used carefully.  Once you do used to it, however, you’ll find it very handy – that method is called 
WAITCNT.   This command, as its name suggests, will wait for the system counter (CNT) to reach a 
specific value.  Until it does the cog executing WAITCNT is put into low-power mode.   
 
The trick with WAITCNT is that it is waiting for a specific value in the system counter, not waiting a given 
duration as with PAUSE.  When using WAITCNT, then, we typically need to involve the current system 
counter value in the target parameter.  For example, if we wanted to create a delay of ¼ second, we can 
do it like this: 
 
  waitcnt(cnt + (clkfreq >> 2)) 
 
As you can see we’re taking the system frequency (clock ticks per second) and shifting it right by two 
(which is a more efficient way to divide by four) to get the number of clock ticks in ¼ second, then adding 
this to the system counter.   This gives us a 250 ms delay but I think you’ll agree that this line of code is 
not terribly obvious to the casual programmer.    

Another possible problem can arise if the target for WAITCNT has just passed.  Since Spin is an 
interpreted language, it takes a bit of time to execute instructions, and if the specified target has passed 
we can end up with a delay of nearly 54 seconds at 80 MHz (because we have to wrap all the way around 
to the intended value).  Zoinks. 
 
With just a little programming we can have the convenience of PAUSE, making our programs—those that 
can tolerate inline delays—easier to read and maintain.  Let’s give that a go. 
 
To create a method to emulate PAUSE we’ll need to know how many clock ticks are in one millisecond; 
this will be used with WAITCNT.   While we could divide CLKFREQ by 1000 in the working part of the 
program, I think a better way is to create a constant.  At the top of my default Spin template I have the 
following code: 
 
CON 
 
  _clkmode = xtal1 + pll16x 
  _xinfreq = 5_000_000 
' _xinfreq = 6_250_000   
 
  CLK_FREQ = ((_clkmode-xtal1)>>6)*_xinfreq 
  MS_001   = CLK_FREQ / 1_000 
 
The first few lines are standard fare, setting the clock mode and input frequency – we see this in almost 
every Spin program.  You’ll note I’ve got my template setup to easily move between 80 and 100 MHz 
projects.  The next section, some clever code from a Propeller forum member whose name escapes me 
(sorry!) calculates the number of clock ticks per millisecond based on the initial settings.  Since constants 
are pre-calculated by the compiler this adds no overhead to program at run time. 
 
Now we can create a method to emulate PAUSE. 
 
PUB pause(ms) | t         
                          
  t := cnt                
  repeat ms               
    waitcnt(t += MS_001)  
 
On entry we immediately capture the value of the system counter – this is the starting point for the delay.  
Next, we drop into a REPEAT loop for the desired number of milliseconds.  In the WAITCNT instruction we 
add the number of ticks per millisecond to the starting value.  What this does for us is account for the time 
used to run the loop and setup the WAITCNT instruction each time through.  Remember, WAITCNT is 
looking for a specific target and the system counter is always running as we’re working our way through 
the code. 
 
If we re-read the system counter in the WAITCNT instruction, several microseconds would be added to 
each loop – for long delays this could become a problem.  By reading CNT before the loop we ensure that 
each time through runs exactly 1 ms.  The Propeller manual has a great description of this process, called 
synchronized delays; we’ll put this strategy to use often. 

How Long? 
About 10 years ago I created an alarm product that used the BASIC Stamp 2 microcontroller.  The 
product didn’t need much in the way of resolution (100 ms per “tick”), but a tick needed to be 100 ms no 
matter what path the program took.  To make this happen I had to measure the various paths in the code 
and pad them to get to a consistent 100 milliseconds per path. 
 
This process is tedious, to say the least.  I used a spare output pin, setting it high (at the start of a code 
section) and low (at the end of a code section) and monitoring it with an oscilloscope to measure code 

execution time.  Boy, was I glad to get that project finished.  It took as long to tune the program timing as 
it did to write the baseline code! 
 
With multiple cogs and the WAITCNT instruction it’s not likely we’d ever go through this process when 
using the Propeller, but we might want to measure a bit of code to check for performance – this is really 
helpful when experimenting with variations on code.  With the Propeller, we don’t need a spare pin or an 
oscilloscope – we can do this with a couple variables and a terminal program (connected to the 
programming port).  The ability to time a code segment requires just a few lines to be added to our 
template. 
 
PUB main | t0, t1 
                                           
  term.start(31, 30, %0000, 115_200) 
  pause(1) 
  term.tx(CLS) 
 
  repeat 
    t0 := cnt 
 
    ' code to test goes here 
 
    t1 := cnt 
    t1 := ||(t1 - t0) 
 
    term.tx(HOME) 
    term.dec(t1) 
    term.tx(CLREOL) 
    pause(1_000) 
 
After starting a serial object so that we can send the timing result to a terminal, the program drops into a 
REPEAT loop where we set t0 to the present system counter.  After this we’ll insert a bit of code to test 
(not yet, though).  After the code we grab the system counter again and put it into variable t1.  Now we 
can take the difference between the two.  Since CNT is a free-running, 32-bit counter we need to use the 
absolute (||) operator on the difference between t1 and t0.  Note, too, that this process should be only 
used on short-term events. 
 
When I run the program on my Propeller Platform I get 368 counts with no code inserted between the t0 
and t1 checkpoints.  Knowing this we can update the calculation as follows: 
 
    t1 := (||(t1 - t0) - 368) #> 0 
 
The subtraction of 368 is obvious; the rest of the line ensures that we do not dip below zero.  Another 
point: since we’re working with system clock ticks the value does not change with frequency, 
 
For fun, I tested the pause method and found that—at 80 MHz (80,000 ticks per ms)—I got a value of 
81,088 for one millisecond.  This means that there is an overhead of 1,088 clock ticks for the call.  At 
80 MHz this about 13.6 microseconds.  This is the time required to setup the instruction, jump to the 
pause method, and then return to the program.   
 
Can we account for this overhead?  Sure; knowing the value we can update the pause method like this: 
 
PUB pause(ms) | t         
                          
  t := cnt - 1088                
  repeat ms               
    waitcnt(t += MS_001) 

Now... this isn’t perfect.  A delay of one millisecond comes back at exactly 80,000 counts (okay, that’s 
perfect), but a 1,000 ms delay comes back at 80,000,048 counts.  Remember, we are dealing with an 
interpreted language; I think that a 0.6-microsecond error on an inline delay of one second is probably 
okay. 

Tick Tock 
I mentioned earlier that my friend Wayne was working on a timer project.  Originally, we was taking the 
high and low cycle times – which could be expressed in seconds, minutes, or hours – and attempting to 
calculate the number of seconds required for the delay.  This caused a couple problems, not the least of 
which was the opportunity to generate an overflow value on very long cycle times. 
 
A simpler method, I proposed, was to create a software real-time-clock that run independently of the main 
program cog.   Here’s the first version; this one works with values assuming normal clock limits (that is, 60 
seconds, 60 minutes, 24 hours). 
 
PRI softrtc | t0                      
                                                            
  t0 := cnt                               
                                           
  repeat                              
    if (Clock[REG_RST] < 0)           
      longfill(@Clock, 0, 6)          
    waitcnt(t0 += MS_001)             
    if (++Clock[REG_MS] == 1_000)     
      Clock[REG_MS]~                  
      if (++Clock[REG_SC] == 60)      
        Clock[REG_SC]~                
        if (++Clock[REG_MN] == 60)    
          Clock[REG_MN]~              
          if (Clock[REG_HR] == 24)    
            Clock[REG_HR]~          
            if (Clock[REG_DY] < posx) 
              ++Clock[REG_DY]         
            else                      
              Clock[REG_DY]~                         
                                           
As you can see, this method works with a global array called Clock; the array includes elements for 
milliseconds, seconds, minutes, hours, days, and a reset flag.  That last register is important.  Since this 
method is going to be running in a separate copy of the interpreter, hence operating independently of our 
main program, we do not want to change the timing registers outside of this method.  Doing so could 
cause a problem where we try to clear the registers while it is half way through an update cycle. 
 
To get around this we’ll use another method – this just sets the flag register which tells the softrtc 
method to clear its timing elements: 
 
PUB reset           
                   
  Clock[REG_RST]~~  
 
This method sets the last register to -1.  As you can see, the first thing the softrtc method does check this 
value; when it’s below zero all registers (including the reset flag) are cleared to zero in one fell swoop 
using LONGFILL.  And as the software RTC uses a synchronized one-millisecond delay, resetting the 
clock is reliable and happens within a millisecond. 
 
The code is really quite easy.  When we reach 1000 milliseconds we clear that register and then add one 
second.  When seconds reaches 60 we clear that register and then add a minute; and so on.   

To launch this method into its own interpreter we’ll do this: 
 
  cognew(softrtc, @Stack) 
 
This form of the COGNEW instruction launches another Spin interpreter into a separate cog and points this 
cog to the softrtc method.  Note that the Spin interpreter needs some stack (RAM) space, and in this 
case we’re telling the cog to use an array of longs called Stack.   Setting the size of the stack array can 
be a little like guessing the weather for next week – you can get close, but until you get there you don’t 
really know. 
 
That stack is used for the storage of local variables, parameters when calling other methods, and 
intermediate values when evaluating expressions.  I tend to set my stack size to 32 longs and have not 
run into any problems, but then, I tend to keep those Spin methods that run in a separate interpreter fairly 
simple.  There is an object included with the Propeller Tool that can be used to check stack size but I tend 
not to use this as its output is serial – I’d rather do that part myself.   
 
So... I created a really simple object based on an example from Propeller guru (and very nice, very helpful 
guy), Phil Pilgrim.  Phil’s idea is that you can fill your stack with a known pattern and then check the stack 
– while your program is running – for changes from that pattern.   Here are the two methods used in my 
stack test object: 
 
PUB fill(pntr, len) 
 
  longfill(pntr, TEST_VAL, len) 
 
 
PUB used(pntr, len) | idx 
 
  idx := len - 1 
 
  repeat len 
    if (long[pntr][idx] <> TEST_VAL) 
      quit 
    else 
      --idx 
 
  return idx + 1 
 
The first method fills the stack with a known test value – and this must be done prior to using that stack 
(or else you will probably crash the program).  The second method determines stack usage working from 
the top to bottom, looking for a change from the test value; when a change is found QUIT is used to 
terminate the REPEAT loop.   The syntax of the test line may look a little complicated – it’s really not bad.  
We can access any area of memory using the implicit long[] array; that value in the first set of brackets 
(pntr) is the base address of our stack, the value in the second set of brackets (idx) is the index into the 
stack. 
 
I whipped up a little demo (see jm_timer_stack.spin) to test my stack requirements.  Both my stack 
checker and the Parallax version reported a stack usage of seven longs for the softrtc method, so the 32 
originally allocated was more than enough. 
 
The softrtc method works great but Wayne pushed back – what he wanted to do is be able to set timing 
units beyond normal clock boundaries; for example, he might want to set the “on” side of the cycle to 90 
seconds (in seconds), and have the “off side of the cycle set to 210 seconds (total cycle time is 300 
seconds; five minutes). 
 
Okay, it’s just SMOP (a small matter of programming).  I copied the softrtc method and created a new 
one called freerun; this version is a free-running timer with very little bounds checking. 

PRI freerun | t0                         
                                                            
  t0  := cnt                                  
                                              
  repeat                                      
    if (Clock[REG_RST] < 0)                   
      longfill(@Clock, 0, 6)                  
    waitcnt(t0 += MS_001)                        
    if (++Clock[REG_MS] == 1_000)             
      Clock[REG_MS]~                          
      if (Clock[REG_SC] < posx)               
        if ((++Clock[REG_SC] // 60) == 0)     
          if ((++Clock[REG_MN] // 60) == 0)   
            if ((++Clock[REG_HR] // 24) == 0) 
              ++Clock[REG_DY]                 
      else                                    
        Clock[REG_RST]~~                      
 
This method is structurally similar to softrtc but we don’t do the same bounds checking on the seconds, 
minutes, and hours registers.  We do of course check the milliseconds register and treat it the same way.  
Seconds, however, has a new boundary: POSX.  This is an internal value that is the largest positive value 
in the Propeller’s 32-bit integer system – it’s just a hair over 2.1 billion (the 32nd bit is the sign bit).  
 
The reason for this limit is that we’re going to use the modulus operator (//) to check for updates to the 
other registers when the seconds register changes.  Since // treats integers as signed, we have to limit 
the seconds register to 31 bits.  Don’t be concerned about this limitation – we could let the timer run for 
68 years before the seconds register forces the timer to reset itself. 
 
When I showed Wayne this code he was happy about the flexibility, but concerned about the accuracy 
with all that math when value changes are cascading – modulus relies on division which is notoriously 
slow.  I did a quick check using the time tester we played with earlier and determined that even under the 
worst case conditions, when every register updated on a given cycle, the whole works took about one-
eighth of a millisecond; that leaves us plenty of time to get back to and be sitting on the WAITCNT for the 
next 1 ms “tick.” 
 
I also checked the stack – no change there, still uses just seven longs.  After I was happy with the 
routines I folded them into an object that I could use in other programs, and adjusted the stack 
requirement down to 16 (still safe, but not wasteful).   You’ll find this code in jm_softtiming.spin. 
 

Are We Done? 
In early May I was wrapping up a program for a commercial product and had a situation where I wanted 
to alert the user of an error condition (using a new variant of my bi-color LED object), but I didn’t want it to 
remain static; after a short period – say two seconds – I wanted the LED to return to the normal program 
state.  As the program is already using several cogs and I wanted to leave the others free for future 
updates and features, what I needed was a way to kill the error LED without using another cog. 
 
Here’s what I came up with: Since my mainline code is running in a loop about every 50 ms, I decided I 
could just set an error timer and decrement it each time through the loop – once it reached zero I would 
return the LED to the normal state. 
 
To keep things easy I decided to use milliseconds as my timing unit; this meant I needed a way to 
determine the elapsed milliseconds since the last check.  You know where this is going, right? – we’ve 
done this with the code timer.  Here’s the method that returns the number of milliseconds elapsed since 
some starting point: 

PUB elapsedms(tstart) 
 
  return ||(cnt - tstart) / MS_001 
 
There is a caveat here: the limit, at 80 MHz, is about 26.8 seconds – if we wait longer than that between 
tstart and the call to elapsedms we’ll get a bogus return value (I ran into this the hard way when trying to 
using this method with another feature in my product).   
 
To use this method I setup two global variables: LedT0 and LedTimer; the first is set to the system 
counter to create the tstart checkpoint, the second is the number of milliseconds I want the LED to be in 
the error state.  
 
Here’s a bit of code that demonstrates the use of elapsedms (see jm_et_demo.spin). 
 
  repeat                                           
    c := term.rxtime(50)                           
    if (c => 0)                                    
      LedT0 := cnt                                 
                                                   
    case c                                         
      "1"   : led.red                              
              LetTimer := 1_000                     
                                                   
      "2"   : led.green                            
              LetTimer := 2_000                     
                                                   
      "3"   : led.yellow                           
              LetTimer := 3_000                     
                                                   
      other : if (c => 0)                          
                led.set2phase(RED,100,YEL,100)  
                LetTimer := 1_000                   
                                                   
      ' check led                                  
                                                   
    if (LedTimer > 0) 
      etms := elapsedms(LedT0) 
      if (etms < LedTimer) 
        LedTimer -= etms 
        LedT0 := cnt 
      else 
        LedTimer := 0 
        led.off                                  
 
The REPEAT loop will run about every 50 ms if no key is pressed; this is controlled by using the rxtime 
method and setting the timeout value to 50.  If a key is pressed (c will be greater than or equal to zero) 
the LedT0 start point is reset and the key is processed; a valid key gets a solid color, a bogus key causes 
the LED to flash.   For all key events the LED timer is set accordingly. 
 
The bottom of the loop is what processes the LED timing.   When LedTimer is greater than 0 (still running) 
we retrieve the elapsed milliseconds since LedT0.  If this value is less than LedTimer it is subtracted and 
the LedT0 start point is reset.  If etms is greater than the timer value the timer and LED state are cleared. 
 
Why did I do it this way instead of simply calling the pause method for 2000 milliseconds?  Well, my 
product has several manual inputs as well as a serial command stream and I did not want them to be 
delayed for an error LED – especially since on of those inputs could correct the condition.  Some of you 
will no doubt point out that this process can result in a timing error of up to 50 milliseconds; you are 
correct.  That said, this is for a visual indicator, so any small variation in the timing will not be a problem.  
This is a case where imperfect timing took precedence over the use of another cog. 

Wrap Up 
Before I close I want to share one last thing having to do with stacks, though nothing to do with time or 
timing.   Many of us come from BASIC where we often use GOTO in order to get from place-to-place in a 
program.  There is no equivalent of GOTO in Spin, and this can lead to troubles if we’re not aware of it. 
 
For example, I’ve seen new Spin programmers set up a program like this: 
 
PUB main 
  ' do something 
  method1 
 
PUB method1 
  ' do something 
  method2 
 
PUB method2 
  ' do something 
  main 
 
 
While – from the BASIC point of view – this seems logical, it creates a serious problem for Spin.  You see, 
when we call a method, Spin expects that we’re going to come back and continue at the next line, like an 
implied GOSUB.  In the above example the program will work for a while, but will ultimately go haywire.  
The reason is that when we call a method the return address (that is, the address of the line just after the 
call) is pushed onto the stack; at the end of the called method we pop that address off the stack and go 
back – of course, this all happens “under the hood.” 
 
Well, in the framework above we are calling methods in a continuous loop so the stack can eventually 
explode, kind of like Mr. Creosote in Monty Python’s The Meaning of Life after than final “wafer thin mint.”  
It was funny on screen, it’s not funny when it happens in our Propeller programs, and can lead to 
frustrating debugging sessions. 
 
The fix is easy: just remember that the call to a method is like a GOSUB and that the program wants to 
come back when that method is finished.  Knowing this we can re-arrange our main method as follows: 
 
PUB main 
 
  repeat 
    ' do something 
    method1 
    method2 
 
… and, of course, remove the calls at the ends of method1 and method2.  
 
I know this is obvious for you seasoned programmers, but having run into this a couple times – usually by 
those migrating from BASIC – I thought it was worth bringing up. 
 
Okay, then, it’s time (yeah, yeah, bad pun) to get out your Propeller and start working with time-based 
events.   And do spend a bit of time in the manual looking at LONGMOVE – you can use this to take a quick 
snap-shot of your timer registers for timed events, perhaps like a Pinewood Derby timer or similar project 
(see jm_key_timer.spin for a simple example). 
 
Until next time, keep spinning and winning with the Propeller. 

Resources 
Jon “JonnyMac” Williams 
jwilliams@efx-tek.com 
 
Parallax, Inc. 
www.parallax.com 
 
Gadget Gangster 
www.gadgetgangster.com 
– Propeller Platform and accessories 
 
BST (Brad’s Spin Tool) 
www.fnarfbargle.com/bst.html 
– multi-platform IDE supports Spin and PropBASIC 
 
PropBASIC  
www.propbasic.com 


