
Lighting Up the Season — Again! 
By Jon Williams 
 
For Nuts & Volts Magazine, Column 9, November 2010 
 
 
I like this time of year. The air is clear (even in Los Angeles!), the mornings are crisp, and the evenings 
are brightened with holiday decorations that illuminate the insides and outsides of homes everywhere. My 
home is somewhat small, so my lighting projects are, too. Small doesn’t mean wimpy, though, and my 
little 12-channel lighting board for the Propeller Platform is designed to be tough enough for applications 
that go way beyond LEDs. 
 

The Power of 12 
A couple months ago I had the pleasure of working with a leading Hollywood special effects designer to 
add lighting to a big prop he built for a client. The prop is a futuristic soldier in a large suit that has rocket 
engines on the back. My friend wanted to use very bright RGB LEDs in the rockets so that they could be 
faded up to bright red, then slowly transition to blue—all while appearing to pulsate and “rumble” as a real 
rocket engine does. 
 
Output control was pretty easy as I’d previously used a TIP125 in a high-side driver circuit for high-power 
LEDs. I simply whipped up a 12-channel board to handle the “rockets” and other lighting tasks, and then 
took advantage of launching multiple Spin cogs in the Propeller to simplify the code for each element of 
the prop. In the end, my friend was very happy—so much so that he was actually making engine sounds 
with his mouth after we got to code adjusted just the way he liked it!  
 
After buttoning up that project I started thinking about some holiday lighting ideas and then I had a 
thought: if I beef up that circuit—just a little—I can use it for other things, too. With 12 channels I could 
control 12 independent devices, or I could dedicate three channels to an RGB LED module, or I could 
dedicate four channels to control a stepper motor, or I could control any combination thereof. As my friend 
John says, “It’s just SMOP” (small matter of programming). 
 
As I was already using a TIP125 which can handle a fair bit of current, I updated the driver circuit (Figure 
1) by adding a diode across the output terminal; this will let the board handle inductive loads, too. The 
real trick, though, was squeezing 12 of these circuits onto a standard Propeller Platform module and 
running power traces on the top and bottom of the PCB to increase current capability of the board. That 
said, the total current should be limited to 5 A and only if power is coming in on TB13 (when power is 
“borrowed” from the Propeller Platform the connections/traces from the P/P to the driver board cannot 
carry that much current).  

Figure 1 
 
Figure 2 shows a completed board (with 12V test LEDs attached). Yes, it’s pretty busy and will take a little 
patience to build; just take your time with the components and start with the shortest (resistors) and work 
your way up to the tallest (TIP125s). As suggested above, TB13 is available when you need to control a 
lot of power. You an also power the Propeller Platform from TB13—there are power jumpers to the VIN 
headers that need to be installed for this. Note, though, that the power switch on the Propeller Platform is 
bypassed when power from TB13 is shared through these jumpers. 
 

 
 

Figure 2 

With the little room I had left I squeezed two simple circuits onto the board. The first, shown in Figure 3, is 
a simple pushbutton and normally open, dry-contact input. As with the output circuit above there is a pull-
down on the IO pin (that is part of a SIP package). The pull-down is especially important on the output 
circuits to hold the outputs off while the Propeller is in reset and the IOs are floating. 
 
 

 
 

Figure 3 
 

 
The other circuit I added is the TTL serial port shown in Figure 4. The 2.2K resistor in the RX line allows 
5V systems to be connected without problems, and this header can be also used with a standard IR 
sensor for remote control (I did this with the “rocket man” prop—it was fun!). 
 
 
 

 
 

Figure 4 

Oh!, before I forget, there is one final note on assembly: the 10K SIPs are intended to go on the bottom 
side of the board (Figure 5); this keeps them clear of wiring entering the terminal blocks. 
 
 

 
 

Figure 5 
 

Electronic Menorah Redux 
With 12 channels of control at our disposal and a holiday season just begging for cool lighting projects 
let’s jump right in—it’s just a small matter of programming!—right? First up is a do-over of a BASIC Stamp 
program that I wrote for my column four years ago: an electronic Menorah.  
 
With the BASIC Stamp version we used a capacitor on the output to smooth a random pulse to the LED 
to [well, sort of] mimic the flame effect. With the power of the Propeller we don’t need to do that any more, 
and we can get more realism to boot. In the October 2010 issue of Nuts & Volts there is an article that 
shows how to simulate candle flames with the Propeller; we’re going to borrow some of that code and 
build the Menorah program from it. 
 
It the BASIC Stamp version we simply turned it on and it would light the candles for the current day 
(which was incremented at each power-on cycle and stored in the EEPROM). This meant that you had to 
light it on the first day and not miss any days in between. For the update I wanted an easy way to select 
the day at random and to allow manual control over lighting each “candle.” 
 
While watching my test LEDs flicker I remembered a program that I wrote in 1995 for a BASIC Stamp 1 
user. He was a pilot and owned a small airstrip. What he asked me to create was a program that could 
count the number of microphone “clicks” coming through his air-to-ground radio system. The clicks count 

would be used to activate a relay to turn on the runway lights. The idea was to look for a specific number 
clicks in a defined time-frame so that standard radio communications would not trip the circuit. 
 
In case you’re wondering, yes, it worked. We used a 555 to one-shot the audio pulse which allowed the 
BASIC Stamp 1 to detect it. With the Propeller we can do the same thing in software. 
 
As the Power 12 module has a button on it this seemed like good way to go; we can start the sequence 
(indicated by a lit Shamash candle) by pressing in the current day (1 to 8), and then subsequent presses 
will light the next candle. By using the secondary presses the ceremonial blessings can be recited 
between each candle. 
 
Author’s Note: I am not Jewish and do not know if an electronic Menorah is valid for the Hanukkah 
ceremony. Please consult with your Rabbi if you have any concerns in this regard. 
 
I actually developed two helper routines for the button input. The first simply debounces a specified input 
pin for a number of milliseconds; if the button is pressed through the entire cycle then it is considered 
good. Here’s that bit of code: 
 
pub getbutton(pin, dbtime) | db, t 
 
  dira[pin] := 0 
  db := 0 
 
  t := cnt 
  repeat dbtime 
    waitcnt(t += MS_001) 
    if (BTN_MODE == ACTIVE_HIGH) 
      db += ina[pin] 
    else 
      db += !ina[pin] & %1 
 
  return (db == dbtime) 
 
We start by forcing the desired pin to input state and clearing a variable that will count the number of 
active cycles through the loop; if the active cycles matches the specified debounce time then the button is 
considered pressed. 
 
Within the loop we wait one millisecond (synchronized, so as not to add too much overheard for long 
debounce periods) and then add the state of the input to the counter. Note that the code can 
accommodate active-high and active-low inputs. In my experience the buttons on a given project tend to 
use the same active state so I set that state (BTN_MODE) in the constants section. 
 
At the end of the loop we return true or false depending on whether or not the button remained pressed 
through the entire cycle. This code is very modular and provides a fixed delay that we will use in the 
following method. 
 
pub getcount(pin, cmax) | count, idle 
 
  count := 0 
  idle := 0 
 
  repeat 
    if (getbutton(pin, 50) == true) 
      repeat 
      until (getbutton(pin, 50) == false) 
      count += 1 
      idle := 0 

    else 
      idle += 50 
 
    if (count == cmax) 
      quit 
 
    if (count > 0) and (idle => 1000) 
      quit 
 
  return count 
 
Again, the purpose of this method is to count the number of presses on the target input pin, up to a 
specified maximum. It uses the getbutton() method to scan the pin and create internal timing. 
 
At the top we clear the button count and idle timer variables. When a button press is detected the code 
waits for it to be released, increments the count, and then resets the idle timer. If there is no button press 
then the idle timer is incremented by 50 as that is the timing used with getbutton(). 
 
There are two ways to return from this method: 1) the maximum button count is reached, or 2) there is a 
non-zero count and the button has been left idle for at least one second (1000 ms). I considered passing 
the maximum idle time as a parameter, but it really didn’t seem to be worth doing. I spent a lot of time 
playing with this method and it works quite well. That said, you may need to fine-tune it for different 
buttons as some are easier to press than others. 
 
And now we can create the working Menorah code. 
 
pub main | day, level 
 
  leds.init(360) 
 
  bytefill(@wicks, 0, 12) 
  cognew(flicker(9), @stack) 
 
At the top we initialize an object called leds—this is a 12-channel BAM driver that is set up especially for 
the Power 12 PCB so the only thing we have to specify is the desired modulation frequency. The next 
step clears the array called wicks that will hold our candle brightness values. Finally, we launch the 
candle flicker generator into its own cog using cognew (this is discussed in detail in the October 2010 
issue). 
 
The next step is to get the day number from the user and then light the Shamash (“helper”) candle. 
 
  day := getcount(TRIG, 8) 
 
  repeat level from 0 to 255 
    wicks[8] := level 
    pause(6) 
 
The current day is entered with the getcount() method which, for a Menorah, is limited to eight. After 
returning from getcount() we light the Shamash candle which is on channel nine (P8) of the Power 12 
board. A six-millisecond delay between each brightness level causes the “candle” to be fully lit in about a 
second and a half. 
 
The next step is to light each candle in order, starting with the current day. A button press is required for 
each candle so that there is time for the recitation of blessings between each. 

  repeat 
    pause(1000) 
    repeat 
    until (getbutton(TRIG, 50) == true) 
    repeat level from 0 to 255 
      wicks[day-1] := level 
      pause(6) 
    if (--day == 0) 
      quit 
 
Within the loop we allow the last candle to “burn” at least one second before scanning the button for a 
new press. When this is detected the candle for the day is “lit” just as we did above. The day indicator is 
then decremented to point to the next candle. When all are lit we can quit the repeat loop. 
 
The last step is to wait for one more press—this time to extinguish the candles. While I found a lot of good 
information regarding the lighting sequence of the Menorah, I never came across anything having to do 
with extinguishing the candles. This routine, then, will very slowly dim all at the same time.  
 
  repeat 
  until (getbutton(TRIG, 250) == true) 
 
  repeat level from 255 to 0 
    repeat day from 8 to 0 
      if (wicks[day] > 0) 
        wicks[day] := level 
    pause(24) 
 
Note that we require a longer button press to extinguish the candles—this is by design to prevent an 
accidental bump of the button from extinguishing them too early. Note, too, that the loop timing for the 
fade out is longer to create a nice, slow fade out. 
 
So there you have it: a Propeller-Powered electronic Menorah. I think you’ll find the code easy to modify if 
you chose, and the candle simulation is far nicer than the old RC circuit. You can find the complete 
program in the downloads package as jm_pp12_menorah.spin. 

The Kwanzaa Kinara 
Before we move on there is another, multi-day holiday celebration that uses a special candelabra: 
Kwanzaa. Those that celebrate Kwanzaa light the Kinara which has seven candles (one for each day). As 
the Kinara candles are lit in ascending order each day the code is a little simpler. 
 
That said, I employed a little trick in this program to show you how to re-map outputs when they are not in 
the desired order. Let’s say, for example, that we built a beautiful electronic Kinara and wired it with the 
outputs connected to each candle, moving left-to-right, in ascending order; like this 
 
1 2 3 4 5 6 7 
 
The problem is that the Kinara candles are lit in this order: 
 
2 4 6 1 7 5 3 
 
In the event we can’t (or don’t want to) rewire the project we can remap the outputs with a simple table in 
Spin. Here’s a little table I created to remap the (zero-indexed) day to the physical channel output on the 
board: 
 
dat 
 
kinara          byte    3, 0, 6, 1, 5, 2, 4  

On the first day (0) we will light the center candle (black,) which is connected to P3. On the second day 
we will light the outermost red candle which is connected to P0. The map should now make sense. 
 
Here’s the core code for the Kinara: 
 
  repeat 
    repeat 
    until (getbutton(TRIG, 100) == true) 
    pause(250) 
    if (getbutton(TRIG, 150) == false) 
      if (day < 7) 
        idx := kinara[day] 
        repeat level from 0 to 255 
          wicks[idx] := level 
          pause(6) 
        ++day 
    else 
      repeat level from 255 to 0 
        repeat day from 6 to 0 
          if (wicks[day] > 0) 
            wicks[day] := level 
        pause(24) 
      quit 
 
At the top we have a little trickery concerning the button input. Since the Kinara candles are lit in 
ascending order and always start with the central candle, the program doesn’t know what day it is—we 
tell it with each new press. What we do, then, is scan the button, wait a bit, then check again (note these 
periods are slightly longer than before). The idea is that a short button press will light the next candle and 
a long button press will extinguish those that are lit. 
 
You can see that we treat the dat table like a simple array, reading the correct output for the day that 
we’re presently lighting. Easy, right? I think so—and definitely easier than rewiring a project. 
 
See jm_pp12_kinara.spin for the complete, commented listing. 
 
For those that are interested in building a fully functional Menorah or Kinara I encourage you to consult 
the October 2010 issue of Nuts & Volts as there are detailed instructions on building “candles” from LEDs 
and items that you can pick up at any hardware store. 

Playing with Protocols 
More and more of those home Christmas displays that I enjoy so much are coming under computer 
control. I think a good chunk of the credit goes to a nifty bit of freeware called Vixen which was developed 
and maintained by my friend, KC Oaks. 
 
If you’re new to this stuff and really want to dive into the fray I suggest that you check out 
www.doityourselfchristmas.com. It’s a very active set of forums with many knowledgeable, highly-
enthusiastic members. 
 
One of the more popular protocols supported by Vixen and a favorite with many DIY Christmas lighting 
enthusiasts is called Renard, which was created by a gentleman named Phil Short. Most of the code on 
the ‘net for the Renard protocol is PIC-based as that’s what Phil uses, though I have seen an 
implementation done in SX/B. 
 
The Renard system is typically setup such that one board receives the packet from an upstream device 
(console or another unit) and uses the bytes intended for this board while retransmitting all others. I 
believe this design was intended to create a plug-and-play, auto-addressing scheme.  

You see, each controller is looking for a specific two-byte sequence at the start of its data: 
 

$7E $80 <dimmer data>  
 
If we had two controllers in the system the we would see something like this from the console: 
 

$7E $80 <aaaaaaaa> $7E $81 <bbbbbbbb> 
 
In the Renard system the packet that follows $80 is consumed locally and not retransmitted. When the 
second address byte is ($81) received by the first controller it is decremented to $80 which enables the 
second packet for the second controller. Downstream of the first controller the stream above becomes: 
 

$7E $80 $7E $80 <bbbbbbbb> 
 
As you can see, the first packet of dimmer values is removed and the second address is decremented 
such that it will be recognized by the second device in the chain. The second device isn’t bothered by the 
missing packet data; the new SYNC byte that follows the address byte resets everything internally. 
 
With the large base of users one can only assume that the system has proven itself. My question 
became, can I create a half-duplex (receive-only) device that would work with the protocol? I think the 
answer is yes and I’m going to show you want I did. 
 
From my view of the world the Renard protocol has three major states: 

 wait for sync byte ($7E) 
 wait for address byte ($8x) 
 process packet bytes 

 
To send the SYNC value as a dimmer level it gets encoded as two bytes: $7F, $30. What this means, 
then, is that we need two-byte encoding for a $7F (ESC) dimmer level—that would be $7F, $31. Finally, 
there is a “pad” byte that may be sent by the console that is ignored by the Renard device. To use the 
PAD value ($7D) in the dimmer packet it gets encoded as $7F, $2F. 
 
Okay, then, let’s give it a whirl. For most serial applications we can use the FullDuplexSerial object to do 
the heavy lifting. Yes, we only need half duplex (reception) for this project but FDS is still just fine. To 
setup FDS we will typically do something like this: 
 
  console.start(RX1, TX1, %0000, 115_200) 
 
The serial object is called “console” and we’re going to receive at 115.2K baud on pin RX1 (the 
programming port to make testing easy). The third parameter is the mode; in this case we’re going with 
the standard true mode which works with RS-485 devices, as well USB-Serial devices like the Prop Plug 
and the USB2SER. 
 
And now for the fun! What I did is setup a simple state handler to accommodate the rules defined for the 
Renard protocol. Again, my code is just a “sniffer” (that is, it doesn’t retransmit anything) so it will be 
looking for an address that could be something higher than $80 (which it must be if there are standard 
Renard devices in use on the same buss). 
 
  console.rxflush  
  state := W_SYNC 
  chan := 0 
 
  repeat 
    cmd := console.rx 
    case state 
      W_SYNC : if (cmd == SYNC) 
                 state := W_ADDR 

This is the basic setup and handling the first state. The receive buffer gets flushed after the serial object is 
started just to make sure that there’s no garbage in it. After initializing the state and chan variables (which 
are local to the method therefore must be manually initialized) we drop into a repeat loop that will 
execute the state handler. 
 
At the top of the loop we receive a byte from the console and drop into a case structure. The first task is 
to look for the SYNC byte. When that happens we can advance the state to look for the address byte that 
matches our board. 
 
      W_ADDR : if (cmd == PAD) 
                 ' ignore 
               elseif (cmd == SYNC) 
                 ' no change 
               elseif (cmd == ADDR) 
                 state := W_LVL1 
                 chan := 0 
               else 
                 state := W_SYNC   
 
In the W_ADDR state the program is waiting for the address byte, but other values could arrive and we 
need to handle them. If a PAD ($7D) byte or another SYNC byte arrives there is no change in state. If the 
correct address byte arrives then the state is bumped to start reading dimmer values from the stream and 
the channel pointer is reset to zero. If any other value arrives we reset the handler and wait for the next 
SYNC byte. 
 
      W_LVL1 : if (cmd == PAD) 
                 ' ignore it 
               elseif (cmd == SYNC) 
                 state := W_ADDR 
               elseif (cmd == ESC) 
                 state := W_LVL2 
               else 
                 leds.set(chan, cmd)  
                 if (++chan == CHANNELS)               
                   state := W_SYNC  
 
In most cases the packet will arrive very cleanly but, as we did above, we need to handle any exceptions 
as well. If a PAD byte comes in we ignore it. If a new SYNC byte shows up then we reset the state 
processor to look for the board address. If an ESC byte arrives then the channel value is encoded as two 
bytes and we update the state handler to process the second byte. Finally, if none of that happens we 
simply take the new byte as the current channel value. After updating the dimmer the channel count is 
incremented, and when we’ve reached the limit for this board the state is reset to start the process for the 
next packet. 
 
For those dimmer values that are encoded with two bytes the W_LVL2 state handles the second byte and 
updating the dimmer channel.  
 
      W_LVL2 : if (cmd => $2F) and (cmd =< $31) 
                 cmd := $7D + (cmd-$2F) 
                 leds.set(chan, cmd) 
                 if (++chan == CHANNELS)       
                   state := W_SYNC 
               elseif (cmd == SYNC) 
                 state := W_ADDR 
               else 
                 state := W_SYNC 

If the second byte is in the correct range ($2F to $31) the dimmer value is calculated and LED channel is 
updated. As above, the channel count is checked and if the value just received is for the last channel, the 
state processor is reset. 
 
If we happen to get a “broken” packet and a new SYNC byte shows up, then we go back to looking for the 
address byte, otherwise we go back to looking for the next SYNC byte. 
 
Whew! I know that reading about that process probably makes it seem more difficult than it is—once you 
see the full listing (in jm_pp12_renard_ez.spin) it will be easier to grasp. Does it work? Absolutely!—and 
I blasted packets at it all day long at 115.2K baud (most Renard systems run at 57.6K or slower). Figure 6 
shows my Vixen test pattern that I sent to the board during testing. Every time I looked in on it the pattern 
was happily running on the LEDs, so the trouble we took to make the state handler robust seems to have 
paid off. 
 

 
 

Figure 6 
 
Vixen supports a lot of interesting protocols and with the flexibility of Spin it will be worth investigating 
more of them—but let’s save that for another time, shall we? Since we could add two half-duplex ports to 
the Propeller Platform using the Gadget Gangster Prototyping board, it might be fun to create a traditional 
Propeller-based Renard device that handles the pass-through. Are you up to it? Of course you are, and 
I’ll probably try it myself down the road.... 
 
For now, though, and as this is my last column of the year, let me wish you and yours the very best for 
this holiday season and the coming New Year. I am grateful to get to share my experiments in Nuts & 
Volts and sincerely appreciate Robin, Larry, and all the nice folks I interact with a T&L Publications. Of 
course, I must also thank my friends at Parallax (that started for me with a BASIC Stamp 1 some 16 years 
ago), my colleague, John Barrowman, and, finally, my wonderfully supportive friends and family. May God 
bless you all. 
 
Until next time, light up your life and keep spinning and winning with the Propeller! 

Resources 
Jon “JonnyMac” Williams 
jwilliams@efx-tek.com 
 
Parallax, Inc. 
www.parallax.com 
 
Vixen 
www.vixenlights.com 
 
Gadget Gangster 
www.gadgetgangster.com 
– Propeller Platform kits and accessories 
 

Bill of Materials 
D1-D12 1N4001           Mouser 512-1N4001 
JMPR1 0.1 SHUNT        Mouser 517-950-00 
Q1-Q12 2N3904           Mouser 610-2N3904 
Q13-Q24 TIP125           Mouser 512-TIP125 
R1-R12 470 ohm          Mouser 299-470-RC 
R13-R24 1K ohm           Mouser 299-1K-RC 
R25 2.2K ohm         Mouser 299-2.2K-RC 
R26  10K ohm          Mouser 299-10K-RC 
RN1-RN12          8x10K, SIP       Mouser 81-RGLD8X103J 
SW1 6mm NO Mouser 653-B3F-1022 
TB1-TB13  0.2 Euro Mouser 571-2828362 
X1-X8            0.1 M-STRT Mouser 517-6111TG* 
PCB  ExpressPCB 

* The 517-6111TG is a 40-pin component; only two are required for the project (split as needed). 
 


