
SIRCS, Propeller Style! 
By Jon Williams 
 
For Nuts & Volts Magazine, Column 4, January 2010 
 
 
I have a really cool friend named Lou who designs camera support accessories, one of those accessories 
being a nice pan/tilt control head for film and video cameras.  I worked with Lou to redesign the 
electronics for this product to use the SX48.  Things are going well and I suggested to Lou that he 
consider the Propeller for the next generation; I felt like Spin would suit him more than SX/B and allow 
him to add features without relying on me or anyone else.  Immediately thereafter Lou called and asked if, 
in a future version of that product, we could add pan/tilt control from a simple TV remote.  "Sure, I know 
how to do that!" I exclaimed – the problem is that, at the time, anyway, I didn't know how to do that with 
the Propeller!  It didn't take long (only an hour or two) to make it work and I'm happy to have SIRCS 
reception and transmission in my Propeller toolbox – now you can, too. 
 

The SIRCS Protocol in Review 
In the [very unlikely] event that you've never seen or heard of the SIRCS (Sony IR Control System) 
protocol I'll explain: SIRCS is a pulse-width modulated protocol transmitted over an IR beam that is 
primarily used in the consumer electronics arena (TVs, VCRs, DVD players, etc.).   
 
On the receiving side of an SIRCS-based system a demodulator provides an active-low output to the 
processor as shown in Figure 1.  The SIRCS stream begins with a Start bit that is 2.4ms wide.  This bit, 
as with all other bits, is followed by a 0.6ms pad.  The data bits, which are transmitted LSB-first, are 
width-encoded with a "1" bit being 1.2ms wide and a "0" bit being 0.6ms wide.  The entire frame is 
transmitted within a 45ms window.  Most remotes will repeat the SIRCS code at least three times to 
ensure it has been received, though I've noted that my Sony still camera remote actually transmits a 
given code five times. 
 

 
Figure 1 

 
Note that different devices use different length codes.  In my experience, TV codes tend to be 12 bits, 
while more advanced devices like DVD players use 20-bit codes.  The IR code has two elements: a 
device code and a key code.  In the 12-bit system the device code is five bits, the key code is seven bits.  
In the 20-bit system the device code is eight bits, the key code is 12 bits.  As you can see, the 20-bit 
system provides for significantly more key codes – this makes sense considering the increasing 
complexity of consumer electronics devices. 

Decoding SIRCS with the Propeller 
The first bit of great news when using the Propeller is that we can drop SIRCS decoding into its own cog 
so we don't have to worry about interrupt timing (which doesn't exist on the Propeller, anyway), a code 
coming in when we're not ready for it, or "blocking" the rest of the program while waiting for the user to 
press a key on the remote. 
 
Let's jump right in, shall we?   You may remember from the DMX512 project that in addition to using a 
separate cog we also took advantage of the multi-purpose timers within each cog to simplify the code.  
We're going to do that again. 
 
rxsircs         mov     ctra, NEG_DETECT 
                mov     frqa, #1 
 
                mov     ctrb, FREE_RUN 
                mov     frqb, #1 
 
waitok          rdlong  tmp1, okpntr    wz 
        if_nz   jmp     #waitok 
 
waitstart       waitpeq irmask, irmask 
                mov     phsa, #0 
                waitpne irmask, irmask 
                mov     phsb, #0 
                waitpeq irmask, irmask 
                cmp     START_BIT, phsa wc 
        if_nc   jmp     #waitstart 
 
As you can see we're going to use both counters; counter A is going to be setup in negative detect mode 
to measure the width of incoming bits, and counter B will be used in free-run mode to keep track of frame 
timing. 
 
At the label waitok the program reads a flag from the hub.  This allows the user to enable the SIRCS 
decoder when desired.  A non-zero value in this flag allows the program to drop through to waitstart 
where, as you've no-doubt guessed, we will wait for the 2.4ms start bit. 
 
At the beginning of this section we can use watipeq and the IR input pin mask to wait for that input to be 
high.  We need to start this process while the line is high to make sure that we can measure the whole bit 
– we don't want to come in late and get an inaccurate measurement.  When the line is high we clear 
PSHA and then wait for the line to drop with waitpne. 
 
When the line goes low we clear the PHSB register.  Why?  Well, we're assuming at this point that we've 
found the leading edge of the start bit and we need to clear the PHSB register which will be keeping track 
of the frame duration.  Once the line goes back high we use cmp to test the width of the bit – if it's a valid 
start bit then we drop through, otherwise we jump back to waitstart and test the next bit. 
 
With the start bit detected the next step to collect the bits that comprise the device/key code. 
 
                mov     irwork, #0 
                mov     bits, #0 
 
checkframe      cmp     MS_044, phsb    wc 
        if_c    jmp     #irdone 
 
waitbit         test    irmask, ina     wz 
        if_nz   jmp     #checkframe 
 
measurebit      mov     phsa, #0 
                waitpeq irmask, irmask 
                cmp     ONE_BIT, phsa   wc 

                rcr     irwork, #1 
 
                add     bits, #1 
                cmp     bits, #20       wc 
        if_b    jmp     #checkframe 
 
irdone          mov     tmp1, #32 
                sub     tmp1, bits 
                shr     irwork, tmp1 
 
report          wrlong  irwork, codepntr 
                wrlong  bits, bitspntr 
                wrlong  DONE, okpntr 
 
                jmp     #waitok 
 
Our decoder has no way of know what's being pointed at it so it keeps track of the code received (irwork) 
as well as the number of bits received (bits) – both values will be provided to the calling program. 
 
It may seem like odd placement, but the top of the bit receive loop (at checkframe) actually tests the 
frame timer (in PHSB) to see if the frame is complete.  I'm using 44 milliseconds as the test value for what 
is supposed to be a 45ms frame.  It's okay that this is a little short as even if all the bits of a 20-bit code 
were "1" the transmission of the start and those bits would only be 39 milliseconds. 
 
If the frame is still active we test the line for a new bit (line is low).  While the line remains high we have to 
loop back through checkframe so that the end is handled correctly.  Once the line does drop for a new bit 
we clear PHSA and then wait for the line to go back high.  The width of the newly-captured bit is 
compared to the timing for a "1" bit with the result written into the Carry flag.  This bit is then moved into 
irwork with rcr (rotate carry right).  With the bit saved we increment the bit count and if less (if_b) than 20 
bits jump back to checkframe. 
 
Once we've received 20 bits or the frame timer reaches 44 milliseconds the program moves to irdone 
where we clean up the result.  Since the bits come in LSB first we had to shift them in from the left 
(toward the right) – this means that our result is MSB-aligned in irwork.   By subtracting the number of bits 
received from 32 (bits in a long) we can correct and LSB-align the result.  One of my favorite aspects of 
PASM is that we can shift a value any number of bits with a single instruction. 
 
Finally, the result in irwork, the number of bits received, and the enable flag are written back to the hub.  
The flag is set to false so the code doesn't run until the calling program tells it to. 

Spinning Up an SIRCS "Sniffer" 
A year a go I created an SX-based SIRCS "sniffer" program to determine the various codes of the Sony 
remotes that seem to run wild in my home – it now makes perfect sense to create a Propeller version of 
that.  The code is really simple.   In addition to the SIRCS receiver object, we'll use a serial output object 
to send the detected SIRCS codes to a terminal program (I prefer the Parallax Serial Terminal for my 
Propeller experiments).   Here's the code: 
 
pub main | code, bc 
 
  ir.init(0) 
  term.init(30, 115_200) 
  waitcnt(clkfreq / 1_000 + cnt) 
 
  term.str(string( { 
  } CLS, "SIRCS Sniffer", CR, CR)) 

  repeat 
    code := ir.getir 
    bc := ir.bitcount 
    case bc 
      12: 
        term.str(string("12 :: ")) 
        term.bin(code >> 7, 5) 
        term.tx(".") 
        term.bin(code, 7) 
      20: 
        term.str(string("20 :: ")) 
        term.bin(code >> 12, 8) 
        term.tx(".") 
        term.bin(code, 12) 
      other: 
        if bc < 10 
          term.tx("0") 
        term.dec(bc) 
        term.str(string(" :: ")) 
        term.bin(code, 32) 
    term.tx(CR) 
    waitcnt(clkfreq / 4 + cnt) 
 
The main loop of this program calls the .getir() method from the IR object.  This method enables the 
SIRCS receiver, waits for a code to show up, and then returns it to the caller.  At this point there is a valid 
bit count available as well and this can be acquired using the .bitcount() method.   
 
As you can see, the bulk of the code has to do with formatting the output so that it looks nice on the 
screen.   The output sections print the bit count and then separate the device and key codes so that 
they're easy to see.  In the event an odd bit count is received (it has never happened to me) the program 
handles this in the other section of the case statement.   At the end of the loop a short delay is inserted 
to prevent overrunning the transmit buffer. 
 
The hardware for this program is a no-brainer: Figure 2 shows the PNA4602M decoder that we've used in 
the past; the addition of the 2.2K resistor is to limit the amount of current flowing through the Propeller's 
IO pin protection diodes (these come into play with the decoder output being 5v).  Figure 3 shows the 
circuit connected on a Propeller Demo Board and, finally, Figure 4 shows the output to a terminal window 
(using PST).  The first two lines represent the channel up and down buttons on my TV/DVD multi-function 
remote; the last two lines represent the volume up and down buttons. 
 
 

 
 

Figure 2 

Figure 3 
 

 
 

Figure 4 
 
 
Experiment with your remotes – I'm sure you'll find, as I have, that even within the same brand (using 
SIRCS) various models will have their own device code, even if the key code matches.  The up-side of 
this is that we can design a program to respond to a very specific brand of remote. 

Transmitting SIRCS with the Propeller 
I got started with SIRCS transmission I needed a way to control my Sony DSLR to create time-lapse 
videos.   What was somewhat challenging using the SX becomes pretty easy in the Propeller, especially 
when we take advantage of one of the counters to handle the IR LED modulation frequency. 
 
Figure 5 shows how to connect an IR LED to the Propeller; note that we're modulating and controlling 
from the cathode end, with the anode being tied to 3.3v.  You may remember that in the SX we used two 
pins: one for modulation, one for control.  So why just one pin when using the Propeller?  What changed? 
 

 
 

Figure 5 
 
 
In the Propeller's architecture any element that can make an output pin go high is OR'd together within 
the cog before being OR'd with other cog signals and routed to the final output (refer to the Propeller 
block diagram in the user manual).  Within every cog there are four elements that can make an output pin 
go high: the output register, counter A, counter B, and the video generator.  
 
What this means for us, then, is that we can modulate the LED by directing a counter set to NCO mode to 
the cathode pin, and disable that signal when desired by writing a "1" to the same pin.  Doing this will hold 
the cathode high and the LED will turn off.  When we take that pin low the modulation signal from the 
counter can past through to the LED.  Pretty neat, huh? 
 
Before jumping into the code there is one small bit of business.  When using a counter in NCO mode we 
need to set the FRQx register to provide the desired output frequency.  The formula for the proper FRQx 
setting in NCO mode is: 
 

FRQx = Hz x 232 / System Frequency 
 
As I tend to run my Propeller projects at 80 MHz I've created constant values for popular modulation 
frequencies used with IR remotes. 
 
Okay, then, let's do it.  On entry we need to make the IR cathode pin an output and high to turn it off, and 
then setup a counter to provide the modulation signal.  
 
txsircs         or      outa, ircath 
                or      dira, ircath 
                mov     frqa, modfreq 
                mov     ctra, modctrl 
 
waitcmd         rdlong  frcount, fcpntr wz 
        if_z    jmp     #waitcmd 
                rdlong  bitcount, bcpntr 
                rdlong  code, irpntr 

Transmission is initiated by writing to shared hub variables that hold the code to send, the number of bits 
to use, and the number of frames to transmit.  At waitcmd the number of SIRCS frames is read from the 
hub and when greater than zero we drop through and read the bit count and the device/key code to send.  
The next step is to send it. 
 
startframe      mov     bcount, txbitcount 
                mov     testmask, #1 
                mov     frametimer, MS_45 
                add     frametimer, cnt 
 
txstart         mov     bittimer, BIT_START 
                call    #txbit 
 
Since we'll typically send more than one frame we make a copy of bitcount and then create a mask that is 
setup for bit zero.  The frame timing is set by loading the number of system counter ticks in 45ms into a 
variable called frametimer and then adding the system counter to that – we'll use this value with waitcnt 
after all the bits have been transmitted to pad the frame to the correct duration. 
 
The first element of an SIRCS frame is the 2.4ms start bit.  To handle this we load the variable bittimer 
with the start bit timing and then call a subroutine called txbit.   In previous projects our PASM code didn't 
need subroutines but they are available, and this program is a good place to use them.  Here's that code: 
 
txbit           add     bittimer, cnt 
                andn    outa, ircath 
                waitcnt bittimer, #0 
                or      outa, ircath 
 
txpad           mov     bittimer, BIT_PAD 
                add     bittimer, cnt 
                waitcnt bittimer, #0 
 
txbit_ret       ret 
 
There's really nothing to this: we synchronize bittimer to the system counter (cnt), enable the LED by 
taking the cathode control pin low, and then waiting for the timer to expire.  When it does the LED is 
turned off and at txpad we insert the 0.6ms pad between bits. 
 
Before we leave this there is one important note.  You can see that the last line of the subroutine has a 
special label: txbit_ret.  The nature of the Propeller's architecture and assembler requires that the final 
line of a PASM subroutine is labeled with the subroutine name that is appended with _ret.  So... what 
about those advanced programs where we create a subroutine with two or more entry points?  No 
problem, we can give the ret line multiple labels like this: 
 
name1_ret 
name2_ret       ret 
 
Note that the ret instruction is on the last line. 
 
Okay, with the start bit out of the way we can transmit the device/key code.   This is handled by a simple 
loop that is controlled using bcount (the copy, so it's okay to modify). 
 
txcmd           test    txcode, mask    wz 
        if_z    mov     bittimer, BIT_0 
        if_nz   mov     bittimer, BIT_1 
                call    #txbit 
                shl     mask, #1 
 
checkdone       djnz    bcount, #txcmd  

One of the things I really like about PASM is conditional statements.  The first line at txcmd uses test to 
determine the state of the current bit.  You'll probably remember that test works just like and except that it 
doesn't modify the destination variable (txcode in this case).  The result of test is written to the Z-flag; 
when the bit is zero the Z-flag will be true, when the bit is clear the Z-flag will be false. 
 
The next two lines take care of setting bittimer to the appropriate value, using the conditional statements.  
The condition clause, if_z, will load bittimer with the timing for a zero bit when the bit is in fact zero.  You 
may be wondering what happens to this line when the bit is one.  The condition clause forces the line to 
effectively become a nop – it does nothing.  No harm, no foul; easy-peezy.   
 
With the timing in place we call txbit, then shift the mask one position to the left for the next bit and, 
finally, update the bit count; if there are more bits to send the program jumps back to txcmd.  Once the 
bits are all sent we drop through to waitframe where we hold the program until the 45ms frame timing 
expires.  
 
waitframe       waitcnt frametimer, #0 
                djnz    frames, #startframe 
 
txdone          wrlong  ZERO, fcpntr 
                jmp     #waitcmd 
 
The last step in the process is writing a zero back to the hub where the frames count is stored; this serves 
to alert the caller that the transmission is finished.   With the hub updated we go back to waitcmd and 
stay there until called on to send another SIRCS device/key code frame. 
 
And there you have it, receiving and sending Sony SIRCS codes with the Propeller.  In the download 
package you'll find a little demo that changes channels on my TV – to use this you'll probably want to run 
the "sniffer" program fist to verify the channel up and down keys. 
 
Want to have a little fun?... toss an IR LED onto a demo board, program it to randomly change channels 
or turn off the TV, and then discreetly place the contraption when your friend won't notice or pay attention 
to it.   Hey, who says that those of us who love electronics are up for a tech-based gag from time-to-time? 
– I certainly am! 
 
Until next time, have fun and keep spinning and winning with the Propeller. 

Resources 
 
Jon Williams 
jwilliams@efx-tek.com 
 
Parallax, Inc. 
www.parallax.com 
 
SIRCS Protocol 
http://www.sbprojects.com/knowledge/ir/sirc.htm 
 

Bill of Materials 
IR1  PNA4601M              Parallax 350-00014 
R1   2.2K                  Mouser 291-2.2K-RC 
IR2  IR LED                Parallax 350-00003 
R2   100 ohm               Mouser 291-100-RC 
PCB  Propeller Demo Board  Parallax 32100 

 


